Глава XIII. Повърхнини интеграли

Ще използваме означенията въведени в глава X.

Нека $S:\begin{cases} x = x(u,v) \\ y = y(u,v) \\ z = z(u,v) \end{cases}$ при $(u,v)\in D$ е гладка повърхнина. Коефициентите на нейната първата квадратична форма означаваме с:

$$E(u,v) = \left(\frac{\partial x(u,v)}{\partial u}\right)^2 + \left(\frac{\partial y(u,v)}{\partial u}\right)^2 + \left(\frac{\partial z(u,v)}{\partial u}\right)^2 = \left\|\vec{r}_u(u,v)\right\|^2;$$

$$G(u,v) = \left(\frac{\partial x(u,v)}{\partial v}\right)^2 + \left(\frac{\partial y(u,v)}{\partial v}\right)^2 + \left(\frac{\partial z(u,v)}{\partial v}\right)^2 = \left\|\vec{r}_v(u,v)\right\|^2 \text{ и}$$

$$F(u,v) = \frac{\partial x(u,v)}{\partial u}\frac{\partial x(u,v)}{\partial v} + \frac{\partial y(u,v)}{\partial u}\frac{\partial y(u,v)}{\partial v} + \frac{\partial z(u,v)}{\partial u}\frac{\partial z(u,v)}{\partial v} = \vec{r}_u(u,v).\vec{r}_v(u,v),$$

където $\vec{r}(u,v) = \left(x(u,v), y(u,v), z(u,v)\right)$.

Освен това използваме означенията: $A(u,v) = \det\begin{pmatrix} \dfrac{\partial y(u,v)}{\partial u} & \dfrac{\partial z(u,v)}{\partial u} \\ \dfrac{\partial y(u,v)}{\partial v} & \dfrac{\partial z(u,v)}{\partial v} \end{pmatrix}$;

$B(u,v) = \det\begin{pmatrix} \dfrac{\partial z(u,v)}{\partial u} & \dfrac{\partial x(u,v)}{\partial u} \\ \dfrac{\partial z(u,v)}{\partial v} & \dfrac{\partial x(u,v)}{\partial v} \end{pmatrix}$ и $C(u,v) = \det\begin{pmatrix} \dfrac{\partial x(u,v)}{\partial u} & \dfrac{\partial y(u,v)}{\partial u} \\ \dfrac{\partial x(u,v)}{\partial v} & \dfrac{\partial y(u,v)}{\partial v} \end{pmatrix}$.

Твърдение 1. Ако функциите $x(u,v)$, $y(u,v)$ и $z(u,v)$ притежават първи частни производни, то са в равенствата:

а) $\vec{r}_u(u,v)\times\vec{r}_v(u,v) = \left(A(u,v), B(u,v), C(u,v)\right)$ и

б) $E(u,v)G(u,v) - F^2(u,v) = A(u,v)^2 + B(u,v)^2 + C(u,v)^2$.

1. *Двустранни и едностранни повърхнини в* R^3.

Пример 1. Повърхнината $S:\begin{cases} x = 1+u \\ y = 1+u+v \\ z = 1+\dfrac{v}{2} \end{cases}$ при

$(u,v)\in\Pi = \left\{(u,v): -1\le u\le 1, -1\le v\le 1\right\}$ е успоредника $ABCD$, където $A\left(0,-1,\frac{1}{2}\right)$,

$B\left(2,1,\dfrac{1}{2}\right)$, $C\left(2,3,\dfrac{3}{2}\right)$ и $D\left(0,1,\dfrac{3}{2}\right)$. Успоредника $ABCD$ лежи върху равнината $\alpha : x - y + 2z = 2$. Въвеждаме векторнозначната функция задаваща повърхнината.

```
In[1]:= r[u_, v_] = {1, 1, 1} + u * {1, 1, 0} + v * {0, 1, 1/2}
```

Out[1]= $\left\{1 + u,\ 1 + u + v,\ 1 + \dfrac{v}{2}\right\}$

Тогава върховете A, B, C и D могат да се изчислят така.

```
In[2]:= {r[-1, -1], r[1, -1], r[1, 1], r[-1, 1]}
```

Out[2]= $\left\{\left\{0, -1, \dfrac{1}{2}\right\}, \left\{2, 1, \dfrac{1}{2}\right\}, \left\{2, 3, \dfrac{3}{2}\right\}, \left\{0, 1, \dfrac{3}{2}\right\}\right\}$

Скицираме повърхнината, като горната страна оцветяваме в жълто, а долната страна в синьо. Маркираме и отсечката CD, която е част от границата на успоредника, а също така и средната отсечка на $ABCD$, която е успоредна на CD.

```
In[3]:= s1 = ParametricPlot3D[r[u, v], {u, -1, 1}, {v, -1, 1}, Mesh → None,
          BoundaryStyle → Black, PlotStyle → FaceForm[Yellow, Blue],
          AxesLabel → {x, y, z}];
        s2 = ParametricPlot3D[r[u, 1], {u, -1, 1}, PlotStyle → Thick];
        s3 = ParametricPlot3D[r[u, 0], {u, -1, 1}, PlotStyle → Thick];
        Show[s1, s2, s3]
```

Out[3]=



Повърхнината $ABCD$ е типичен пример на повърхнина която има две страни и една граница. (В случая границата е частично гладка.) Горната страна е оцветена в жълто, а долната - в синьо. Сега нека да огънем правоъгълника така, че да се отъждестви ("залепи") отсечката BC с отсечката AD, така че точката A да съвпада с точката B и точката D да съвпада с точката C. Получава се повърхнината P, която е част от цилиндрична повърхнина. Параметрично тази повърхнина може да се зададе с равенствата

$$S:\begin{cases} x=\dfrac{\pi}{\sqrt{2}}\cos(u) \\ y=\dfrac{\pi}{\sqrt{2}}\sin(u) \\ z=v \end{cases} \text{ при } (u,v)\in\Pi=\left\{(u,v): 0\le u\le 2\pi, -\frac{\sqrt{5}}{2}\le v\le\frac{\sqrt{5}}{2}\right\} \text{ и}$$

има вида

```
In[4]:= r1[u_, v_] = {π/√2 * Cos[u], π/√2 * Sin[u], v};

p1 = ParametricPlot3D[r1[u, v], {u, 0, 2 Pi}, {v, -√5/2, √5/2},
   Mesh → None, BoundaryStyle → Black, PlotStyle → FaceForm[Yellow, Blue],
   AxesLabel → {x, y, z}];

p2 = ParametricPlot3D[{π/√2 * Cos[u], π/√2 * Sin[u], √5/2}, {u, 0, 4 Pi},
   PlotStyle → Thick];
p3 = ParametricPlot3D[{π/√2 * Cos[u], π/√2 * Sin[u], 0}, {u, 0, 4 Pi},
   PlotStyle → Thick];
Show[p1, p2, p3]
```

Out[4]=



На чертежа се вижда ясно, че повърхнината P има външна страна, оцветена в жълто и вътрешна страна оцветена в синьо. Можем да кажем, че при отъждествяването на

отсечката BC с отсечката AD долната страна на успоредника $ABCD$ се е превърнала във вътрешната страна на повърхнината P, а горната страна – във външната страна на P. Освен това отсечка CD се е превърнала в горната част на границата на P. Маркирано е и мястото на залепване.
Сега нека да огънем правоъгълника $ABCD$ така, че да се отъждествят ("залепят") отсечките BC и AD, но за разлика от горния случай сега точката A да съвпада с точката C и точката D да съвпада с точката B. Получава се повърхнината G, която е известна под името лист на Мьобиус. Параметрично такава повърхнина може да се зададе с равенствата $S:\begin{cases} x=\left(1+v\cos\left(\dfrac{u}{2}\right)\right).\cos(u) \\ y=\left(1+v\cos\left(\dfrac{u}{2}\right)\right).\sin(u) \\ z=v.\sin\left(\dfrac{u}{2}\right) \end{cases}$ при $(u,v)\in\Pi=\{(u,v):0\le u\le 2\pi,-1\le v\le 1\}$ и има вида

```
In[5]:= r2[u_, v_] = {(1 + v * Cos[u/2]) * Cos[u], (1 + v * Cos[u/2]) * Sin[u], v * Sin[u/2]};
        ParametricPlot3D[r2[u, v], {u, 0, 2 * Pi}, {v, -1, 1},
         PlotStyle → FaceForm[Yellow, Blue], AxesLabel → {x, y, z}]
```

Out[5]=


Ако разгледате чертежа от различен ъгъл, ще видите как синята и жълтата страна са се слели и се е получила повърхнина с една страна и един ръб.

```
In[6]:= g1 = ParametricPlot3D[r2[u, v], {u, 0, 2*Pi}, {v, -1, 1}, Mesh → None,
          AxesLabel → {x, y, z}, AxesLabel → {x, y, z}];
        g2 = ParametricPlot3D[r2[u, 1], {u, 0, 4 Pi}, PlotStyle → Thick,
          AxesLabel → {x, y, z}];
        g3 = ParametricPlot3D[r2[u, 0], {u, 0, 4 Pi}, PlotStyle → Thick,
          AxesLabel → {x, y, z}];
        Show[g1, g2, g3]
```

Out[6]=



Нека сега S е произволната гладка повърхнина въведена в началото на главата. Върху S определяме *векторно поле от нормали*, като на всяка точка $(x, y, z) \in S$ съпоставим единичен вектор на нормалата $\vec{n}(x, y, z)$. За всяка точка има точно два такива вектора. Например, ако $(x, y, z) = \vec{r}(u, v)$, то можем да положим $\vec{n}(x, y, z) = \dfrac{\vec{r}_u(u, v) \times \vec{r}_v(u, v)}{\left\|\vec{r}_u(u, v) \times \vec{r}_v(u, v)\right\|}$ или $\vec{n}(x, y, z) = -\dfrac{\vec{r}_u(u, v) \times \vec{r}_v(u, v)}{\left\|\vec{r}_u(u, v) \times \vec{r}_v(u, v)\right\|}$. За определеност считаме, че $\vec{n}(x, y, z) = \dfrac{\vec{r}_u(u, v) \times \vec{r}_v(u, v)}{\left\|\vec{r}_u(u, v) \times \vec{r}_v(u, v)\right\|}$. Понеже повърхнината е гладка, то определеното по този начин векторно поле от нормали $\vec{n}(x, y, z)$ е непрекъснато в малка околност на всяка точка (x, y, z).

Дефиниция 1. Повърхнината S се нарича *двустранна,* когато върху нея може да се определи векторно поле от нормали непрекъснато върху цялата повърхнина S.
В противен случай повърхнината се нарича *едностранна*.
Като синоними се употребяват: *ориентируема* повърхнина вместо *двустранна* повърхнина и *неориентируема* повърхнина вместо *едностранна* повърхнина.

Пример 2. Дефинираме функцията $F(R)$, която изчислява единичния нормален вектор $\vec{n}(u,v)=\dfrac{R_u(u,v)\times R_v(u,v)}{\left\|R_u(u,v)\times R_v(u,v)\right\|}$, където $R(u,v)=\left(x(u,v),y(u,v),z(u,v)\right)$.

```
In[7]:= F[R_] := Module[{n, n1, n2},
          n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
          n2[u_, v_] := n[u, v].n[u, v] // Simplify; n1[u_, v_] := n[u, v]/Sqrt[n2[u, v]]; n1[u, v]]
```

Правоъгълника $ABCD$ от пример 1 се задава с $R=r(u,v)=\left(1+u,1+u+v,1+\dfrac{v}{2}\right)$ при $(u,v)\in\Pi=\left\{(u,v):-1\le u\le 1,-1\le v\le 1\right\}$. Тогава, използвайки дефинираната в пример 1 функция $r(u,v)$ получаваме за $\vec{n}(x,y,z)$

```
In[8]:= F[r]
```

$$\text{Out[8]= } \left\{\frac{1}{\sqrt{6}},-\frac{1}{\sqrt{6}},\sqrt{\frac{2}{3}}\right\}$$

Следователно $\vec{n}(x,y,z)\equiv\left(\dfrac{1}{\sqrt{6}},-\dfrac{1}{\sqrt{6}},\sqrt{\dfrac{2}{3}}\right)$ и е непрекъсната върху $ABCD$.

Избирането на това векторно поле е еквивалентно на избирането на горната страна на $ABCD$. В пример 1 тази страна е оцветена в жълто. Ще отбележим, че избора на ориентация върху $ABCD$ е еквивалентно и на избор на определена посока върху неговата граница. Обхождането на границата трябва да е обратно на часовниковата стрелка когато се гледа от избраната страна. В случая това е посоката $A\to B\to C\to D\to A$. На чертежа това изглежда така

```
In[10]:= a1 = {0.12, 0.12, 0.1}; b1 = {-0.12, 0.12, 0.01}; c1 = {0.12, -0.12, -0.1};
         d1 = {0, 0.1, -0.2};
         s4 = ListPointPlot3D[{a, b, c, d}, PlotStyle → PointSize[Large]];
         s5 = Graphics3D[{{Thick, Arrow[{a, b}]}, {Thick, Arrow[{b, c}]},
             {Thick, Arrow[{c, d}]}, {Thick, Arrow[{d, a}]}, Text["A", a + a1],
             Text["B", b + b1], Text["C", c + c1], Text["D", d + d1]}];
         Show[s1, s4, s5]
```



Аналогично избора на обхождане на границата $A \to D \to C \to B \to A$ отговаря на избор на векторно поле $\vec{n}_1(x, y, z) \equiv \left(-\frac{1}{\sqrt{6}}, \frac{1}{\sqrt{6}}, -\sqrt{\frac{2}{3}}\right)$ или с други думи отговаря на избора на долната страна на $ABCD$, защото от там погледната обиколката $A \to D \to C \to B \to A$ е обратна на движението на часовниковата стрелка.

Нека сега да разгледаме гладка повърхнина определена като графика на $z = f(x, y)$, $(x, y) \in D$. Намираме $\vec{n}(x, y, z)$

```
In[9]:= R1[u_, v_] = {u, v, f[u, v]}; F[R1]
```

$$\text{Out[9]=}\ \left\{-\frac{f^{(1,0)}[u,v]}{\sqrt{1+f^{(0,1)}[u,v]^2+f^{(1,0)}[u,v]^2}},\ -\frac{f^{(0,1)}[u,v]}{\sqrt{1+f^{(0,1)}[u,v]^2+f^{(1,0)}[u,v]^2}},\ \frac{1}{\sqrt{1+f^{(0,1)}[u,v]^2+f^{(1,0)}[u,v]^2}}\right\}$$

Следователно

$$\vec{n}(x, y, z) = \left(-\frac{f_x(x,y)}{\sqrt{1+\left(f_x(x,y)\right)^2+\left(f_y(x,y)\right)^2}}, -\frac{f_y(x,y)}{\sqrt{1+\left(f_x(x,y)\right)^2+\left(f_y(x,y)\right)^2}}, 1\right)$$ и е непрекъсната върху D.

Понятието ориентация се обобщава и за частично гладки повърхнини. Тези повърхнини можем да разглеждаме като получени след залепването на парчета, които са гладки повърхнини. Такава е например повърхнината на куба. При него едната страна е външната, а другата страна е вътрешната за куба. В случая избора на ориентация върху гладките парчета трябва да е съгласувана така, че ако върху едно парче нормалите сочат навън, то и във всички останали парчета избраните нормали трябва да сочат навън. Точно такова съгласуване не е възможно да се направи при

Мьобиусовия лист и той е едностранна повърхнина. *За едностранните повърхнини е характерно, че има затворен контур при обиколката на който нормалата се изменя така, че стига до противоположната.* Ще илюстрираме казаното, като разгледаме върху Мьобиусовия лист от пример 1 затворения контур $\gamma_0 : r_2(u,0) = (\cos(u), \sin(u), 0)$ за $0 \le u \le 2\pi$. Използваме определената в пример 1 функция $r_2[u, v]$. Действително

```
In[10]:= r2[u, 0]

Out[10]= {Cos[u], Sin[u], 0}
```

Този контур е маркиран на чертежа на Мьобиусовия лист и можем да го скицираме отделно.

```
In[11]:= ParametricPlot3D[r2[u, 0], {u, 0, 2*Pi}, PlotStyle → Thick,
          AxesLabel → {x, y, z}]
```

Out[11]=



Ясно е, че точката $A(1, 0, 0)$ се получава при $u = 0$ и $u = 2\pi$.

```
In[12]:= {r2[0, 0], r2[2*Pi, 0]}

Out[12]= {{1, 0, 0}, {1, 0, 0}}
```

Намираме единичната нормала в точката $A(1, 0, 0)$

```
In[13]:= n[u_, v_] = F[r2]; n[0, 0]

Out[13]= {0, 0, -1}
```

Сега намираме границата $\lim\limits_{u \to 2\pi -} n(u,0)$

```
In[15]:= Limit[n[u, 0], u → 2*Pi, Direction → 1]

Out[15]= {0, 0, 1}
```

Установяваме, че $\vec{n}(x, y, 0)$ тръгва от точката $A(1, 0, 0)$ като $\vec{n}(1, 0, 0) = (0, 0, -1)$ и движейки се по линията γ_0 се връща в точката $A(1, 0, 0)$ като $\vec{n}(1, 0, 0) = (0, 0, 1)$.

2. *Дефиниция на повърхнинен интеграл*

Нека сега повърхнина S определена с равенствата $(x, y, z) = r(u, v) = (x(u, v), y(u, v), z(u, v))$ за $(u, v) \in D$ и е гладка, двустранна, ограничена и пълна. (*Повърхнината се нарича пълна, когато всяка сходяща редица от точки на S има граница точка от повърхнината.*) Предполагаме, че върху повърхнината S са дефинирани непрекъснатите функции $f(x, y, z)$, $P(x, y, z)$, $Q(x, y, z)$ и $R(x, y, z)$.

С помощта на мрежа от гладки линии разделяме S на краен брой части S_j, $j \in \{1, 2, ..., k\}$. Всяка от частите S_j е гладка, двустранна, ограничена и пълна повърхнина. Като $S = \bigcup_{j=1}^{j=k} S_j$ и различните части S_j нямат общи вътрешни точки.

Диаметър на това разделяне наричаме числото

$$d = \max_{1 \le j \le k} \left\{ \max \left\{ \left\| (x_1, y_1, z_1) - (x_2, y_2, z_2) \right\| ; (x_1, y_1, z_1) \in S_j, (x_2, y_2, z_2) \in S_j \right\} \right\}.$$

Върху всяка една от частите S_j фиксираме по произволен начин точката (x_j, y_j, z_j).

За произволно разделяне и избрани точки (x_j, y_j, z_j) определяме римановите суми

$$\rho_1 = \sum_{j=1}^{k} f(x_j, y_j, z_j) |S_j|, \ \rho_2 = \sum_{j=1}^{k} P(x_j, y_j, z_j) \alpha_j |S_j|, \ \rho_3 = \sum_{j=1}^{k} Q(x_j, y_j, z_j) \beta_j |S_j| \text{ и}$$

$$\rho_4 = \sum_{j=1}^{k} R(x_j, y_j, z_j) \gamma_j |S_j|,$$

където $|S_j|$ е лицето на S_j, $(\alpha_j, \beta_j, \gamma_j) = \dfrac{r_u(u_j, v_j) \times r_v(u_j, v_j)}{\left\| r_u(u_j, v_j) \times r_v(u_j, v_j) \right\|}$ и $(x_j, y_j, z_j) = r(u_j, v_j)$.

Вектора $(\alpha_j, \beta_j, \gamma_j)$ е единичния нормален вектор в точката (x_j, y_j, z_j) на избраната страна на S_j.

Дефиниция 2. Числото I_n е *граница на римановата сума* ρ_n, $n \in \{1, 2, 3, 4\}$, при диаметър на разделянето d клонящ към нула, когато за всяко $\varepsilon > 0$ съществува такова число δ, че за произволно разделяне с диаметър $d < \delta$ и произволни междинни точки е в сила неравенството $|\rho_n - I_n| < \varepsilon$. В този случай пишем $\lim\limits_{d \to 0} \rho_n = I_n$.

Числото $\lim_{d\to 0}\rho_1 = I_1$, когато то съществува, се нарича *повърхнинен интеграл от първи род* и се означава $\iint_S f(x,y,z)\,ds$.

Числото $\lim_{d\to 0}\rho_n = I_n$, за $n>1$ наричаме *повърхнинен интеграл от втори род*.

Използваме означенията $\lim_{d\to 0}\sum_{j=1}^{k} P(x_j,y_j,z_j)\alpha_j\left|S_j\right| = \iint_S P(x,y,z)\,dydz$,

$\lim_{d\to 0}\sum_{j=1}^{k} Q(x_j,y_j,z_j)\beta_j\left|S_j\right| = \iint_S Q(x,y,z)\,dzdx$ и

$\lim_{d\to 0}\sum_{j=1}^{k} R(x_j,y_j,z_j)\gamma_j\left|S_j\right| = \iint_S R(x,y,z)\,dxdy$. Ако и трите повърхнинни интеграла от втори род съществуват, то тяхната сума

$\iint_S P(x,y,z)\,dydz + \iint_S Q(x,y,z)\,dzdx + \iint_S R(x,y,z)\,dxdy$ се нарича *пълен повърхнинен интеграл от втори род* и се означава така

$\iint_S (A,n)\,ds = \iint_S P(x,y,z)\,dydz + Q(x,y,z)\,dzdx + R(x,y,z)\,dxdy$,

където $F(x,y,z) = \big(P(x,y,z),\, Q(x,y,z),\, R(x,y,z)\big)$ е непрекъснато векторно поле върху S, $n(x,y,z)$ е избраното поле от единични нормали върху S.

Свойство 1. От дефиницията на повърхнинните интеграли следва, че повърхнините интеграли от първи род не зависят от избора на ориентация върху S, докато повърхнинните интеграли от втори род зависят от избраната ориентация. По-точно

$$\iint_{S^+} P(x,y,z)\,dydz + Q(x,y,z)\,dzdx + R(x,y,z)\,dxdy = -\iint_{S^-} P(x,y,z)\,dydz + Q(x,y,z)\,dzdx + R(x,y,z)\,dxdy$$

,

където S^+ е повърхнината S с избрана една от ориентациите, а S^- е повърхнината S с избрана една другата ориентация.

Свойство 2. За повърхнинните интеграли е в сила свойството линейност от диференциалната форма по която се интегрира.

Свойство 3. За повърхнинните интеграли е в сила свойството адитивност по повърхнината на интегриране.

Дефиницията на повърхнинни интеграли се обобщава по естествен начин и за частично гладки, двустранни повърхнини като сума на интегралите по съответните гладки парчета. При това обобщение се запазват горните свойства.

3. *Свеждане на повърхнинния интеграл до двоен интеграл*

Теорема 1. Нека сега повърхнина S определена с равенствата $(x,y,z) = r(u,v) = \big(x(u,v), y(u,v), z(u,v)\big)$ за $(u,v)\in D$ и е гладка, двустранна, ограничена и пълна. Предполагаме, че функции $f(x,y,z)$, $P(x,y,z)$, $Q(x,y,z)$ и $R(x,y,z)$ са непрекъснати върху S. Тогава повърхнинните интеграли съществуват и са в сила равенствата:

(1) $\iint\limits_S f(x,y,z)\,ds = \iint\limits_D f\left(x(u,v),y(u,v),z(u,v)\right)\sqrt{E(u,v)G(u,v)-F^2(u,v)}\,dudv$;

(2) $\iint\limits_S P(x,y,z)\,dydz = \iint\limits_D P\left(x(u,v),y(u,v),z(u,v)\right)A(u,v)\,dudv$;

(3) $\iint\limits_S Q(x,y,z)\,dzdx = \iint\limits_D Q\left(x(u,v),y(u,v),z(u,v)\right)B(u,v)\,dudv$;

(4) $\iint\limits_S R(x,y,z)\,dxdy = \iint\limits_D R\left(x(u,v),y(u,v),z(u,v)\right)C(u,v)\,dudv$.

Теорема 1 свежда изчисляването на повърхнинни интеграли до пресмятане на двойни интеграли.

Следствие 1. При условията на теоремата предполагаме, че S е графика на функцията $z = Z(x,y)$, $(x,y) \in D$. Ще намерим запис на формулите (1), (2), (3) и (4) в този случай.

Решение. За произволна S дефинираме функциите:

а) $f_1(u,v) = f\left(x(u,v),y(u,v),z(u,v)\right)\sqrt{E(u,v)G(u,v)-F^2(u,v)}$

```
In[1]:= f1[f_, x_, y_, z_] :=
  Module[{n, n1, n2, R}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := √Simplify[n[u, v].n[u, v]] // PowerExpand;
   n1[u_, v_] := f[x[u, v], y[u, v], z[u, v]] * n2[u, v] // Simplify; n1[u, v]]
```

б) $f_2(u,v) = P\left(x(u,v),y(u,v),z(u,v)\right)A(u,v)$

```
In[2]:= f2[p_, x_, y_, z_] :=
  Module[{n, n1, n2, R}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := n[u, v][[1]];
   n1[u_, v_] := p[x[u, v], y[u, v], z[u, v]] * n2[u, v] // Simplify; n1[u, v]]
```

в) $f_3(u,v) = Q\left(x(u,v),y(u,v),z(u,v)\right)B(u,v)$

```
In[3]:= f3[q_, x_, y_, z_] :=
  Module[{n, n1, n2, R}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := n[u, v][[2]];
   n1[u_, v_] := q[x[u, v], y[u, v], z[u, v]] * n2[u, v] // Simplify; n1[u, v]]
```

г) $f_4(u,v)=R\big(x(u,v),y(u,v),z(u,v)\big)C(u,v)$

```
In[4]:= f4[r_, x_, y_, z_] :=
         Module[{n, n1, n2, R}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
          n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
          n2[u_, v_] := n[u, v][[3]];
          n1[u_, v_] := r[x[u, v], y[u, v], z[u, v]] * n2[u, v] // Simplify; n1[u, v]]
```

Тестваме дефинираните функции с един пример. Нека $r(u,v)=(u,v,uv)$. Тогава $r_u(u,v)=(1,0,v)$, $r_v(u,v)=(0,1,u)$ и

$$r_u(u,v)\times r_v(u,v)=(1,0,v)\times(0,1,u)=\begin{vmatrix} i & j & k \\ 1 & 0 & v \\ 0 & 1 & u \end{vmatrix}=-v.i-u.j+1.k$$

Следователно: $f_1(u,v)=f(u,v,u.v)\sqrt{1+u^2+v^2}$; $f_2(u,v)=P(u,v,uv)(-v)$; $f_3(u,v)=Q(u,v,uv)(-u)$ и $f_4(u,v)=R(u,v,uv)$.

Решаваме същата задача с дефинираните по-горе функции.

```
In[5]:= x[u_, v_] := u; y[u_, v_] := v; z[u_, v_] := u * v; f1[f, x, y, z]

Out[5]= Sqrt[1 + u^2 + v^2] f[u, v, u v]

In[6]:= f2[p, x, y, z]

Out[6]= -v p[u, v, u v]

In[7]:= f3[q, x, y, z]

Out[7]= -u q[u, v, u v]

In[8]:= f4[r, x, y, z]

Out[8]= r[u, v, u v]
```

Получават се същите резултати. *Валидация на дефинираните функции може да се направи и с други решени задачи.* Сега не е проблем да запишем (1), (2), (3) и (4) когато S е графика на функцията $z=Z(x,y)$, $(x,y)\in D$. Пресмятаме $f_1(u,v)$ в общия случай

```
In[9]:= x[u_, v_] := u; y[u_, v_] := v; z[u_, v_] := Z[u, v]; f1[f, x, y, z]

Out[9]= f[u, v, Z[u, v]] Sqrt[1 + Z^(0,1)[u, v]^2 + Z^(1,0)[u, v]^2]
```

Следователно равенство (1) се редуцира до

$$(5)\quad \iint\limits_S f(x,y,z)\,ds=\iint\limits_D f\big(x,y,Z(x,y)\big)\sqrt{1+\left(\frac{\partial Z(x,y)}{\partial x}\right)^2+\left(\frac{\partial Z(x,y)}{\partial y}\right)^2}\,dxdy\,.$$

Пресмятаме $f_2(u,v)$ в общия случай

```
In[10]:= f2[p, x, y, z]
Out[10]= -p[u, v, Z[u, v]] Z^(1,0)[u, v]
```

Следователно равенство (2) се редуцира до

(6) $\iint\limits_S P(x,y,z)\,dydz = -\iint\limits_D P(x,y,Z(x,y))\dfrac{\partial Z(x,y)}{\partial x}dxdy$.

Пресмятаме $f_3(u,v)$ в общия случай

```
In[11]:= f3[q, x, y, z]
Out[11]= -q[u, v, Z[u, v]] Z^(0,1)[u, v]
```

Следователно равенство (3) се редуцира до

(7) $\iint\limits_S Q(x,y,z)\,dzdx = -\iint\limits_D Q(x,y,Z(x,y))\dfrac{\partial Z(x,y)}{\partial y}dxdy$.

Пресмятаме $f_4(u,v)$ в общия случай

```
In[12]:= f4[r, x, y, z]
Out[12]= r[u, v, Z[u, v]]
```

Следователно равенство (4) се редуцира до

(8) $\iint\limits_S R(x,y,z)dxdy = \iint\limits_D R(x,y,Z(x,y))dxdy$.

Следствие 2. а) При условията на теорема 1 за пълния повърхнинен интеграл от втори род е в сила равенството

$$\iint\limits_S (F,n)\,ds = \iint\limits_S P(x,y,z)\,dydz + Q(x,y,z)\,dzdx + R(x,y,z)\,dxdy = \iint\limits_D f_0(u,v)\,dudv ,$$

където $f_0(u,v) = f_2(u,v) + f_3(u,v) + f_4(u,v) = P(x(u,v),y(u,v),z(u,v))A(u,v) +$
$+Q(x(u,v),y(u,v),z(u,v))B(u,v) + R(x(u,v),y(u,v),z(u,v))C(u,v)$.

б) При условията на следствие 1 за пълния интеграл от втори род получаваме равенството

(9)

$$\iint\limits_S (F,n)\,ds = \iint\limits_D \left(-P(x,y,Z(x,y))\frac{\partial Z(x,y)}{\partial x} - Q(x,y,Z(x,y))\frac{\partial Z(x,y)}{\partial y} + R(x,y,Z(x,y))\right)dxdy$$

Това дава възможност, когато се изчислява пълен повърхнинен интеграл от втори род да се използват дефинираните в следствие 1 функции $fj[...]$, $j \in \{2,3,4\}$

Забележка 1. При решаването на следващите примери, ние ще използваме дефинираните в следствие 1 функции $fj[...]$, $j \in \{1,2,3,4\}$ или с други думи теорема 1.

Пример 3. Нека S е частта на хиперболоичния параболоид $z = xy$ изрязана от цилиндричната повърхнина $x^2 + y^2 = a^2$, $a > 0$. Ще пресметнем повърхнинния интеграл от първи род $\iint_S zds$. Намираме се в обхвата на следствие 1 когато $D = \{(x, y) : x^2 + y^2 \le a^2\}$ и $z = xy$. Намираме подинтегралната функция, като използваме дефинираната функция f_1

```
In[13]:= x[u_, v_] := u; y[u_, v_] := v; z[u_, v_] := u*v; f[x_, y_, z_] = z;
         f1[f, x, y, z]
```

Out[13]= $u v \sqrt{1 + u^2 + v^2}$

Следователно $\iint_S zds = \iint_D xy\sqrt{1 + x^2 + y^2}dxdy$. Аналогично на пример 1 от глава 12 намираме интеграла

```
In[14]:= Integrate[%*Boole[u^2 + v^2 < a^2], {u, -a, a}, {v, -a, a}, Assumptions → a > 0]
```

Out[14]= 0

Оказва се, че $\iint_S zds = 0$.

Пример 4. Нека S е частта на параболоида $z = x^2 + y^2$ изрязана от цилиндричната повърхнина $x^2 + y^2 = a^2$, $a > 0$.

1) Ще пресметнем повърхнинния интеграл от първи род $\iint_S zds$. Намираме подинтегралната функция

```
In[24]:= x[u_, v_] := u; y[u_, v_] := v; z[u_, v_] := u^2 + v^2; f[x_, y_, z_] = z;
         f1[f, x, y, z]
```

Out[24]= $(u^2 + v^2)\sqrt{1 + 4u^2 + 4v^2}$

Следователно $\iint_S zds$ е равен на

```
In[25]:= Integrate[%*Boole[u^2 + v^2 < a^2], {u, -a, a}, {v, -a, a}, Assumptions → a > 0]
```

Out[25]= $$\frac{\left(-1 - 2a^2 + 32a^4 + 96a^6 + \sqrt{1 + 4a^2}\right)\pi}{60\sqrt{1 + 4a^2}}$$

2) Ще пресметнем повърхнинния интеграл от втори род $\iint_S zdydz$. Намираме подинтегралната функция

```
In[26]:= f2[f, x, y, z]
Out[26]= -2 u (u^2 + v^2)
```

Следователно $\iint\limits_S zdydz$ е равен на

```
In[27]:= Integrate[% * Boole[u^2 + v^2 < a^2], {u, -a, a}, {v, -a, a}, Assumptions → a > 0]
Out[27]= 0
```

3) Ще пресметнем повърхнинния интеграл от втори род $\iint\limits_S zdzdx$. Намираме подинтегралната функция

```
In[28]:= f3[f, x, y, z]
Out[28]= -2 v (u^2 + v^2)
```

Следователно $\iint\limits_S zdzdx$ е равен на

```
In[29]:= Integrate[% * Boole[u^2 + v^2 < a^2], {u, -a, a}, {v, -a, a}, Assumptions → a > 0]
Out[29]= 0
```

4) Ще пресметнем повърхнинния интеграл от втори род $\iint\limits_S zdxdy$. Намираме подинтегралната функция

```
In[30]:= f4[f, x, y, z]
Out[30]= u^2 + v^2
```

Следователно $\iint\limits_S zdxdy$ е равен на

```
In[31]:= Integrate[% * Boole[u^2 + v^2 < a^2], {u, -a, a}, {v, -a, a}, Assumptions → a > 0]
Out[31]= a^4 π / 2
```

Пример 5. Нека S е частта на конуса $z = b\sqrt{x^2 + y^2}$, $b > 0$ изрязана от цилиндричната повърхнина $x^2 + y^2 - 2ax = 0$, $a > 0$.

1) Ще пресметнем повърхнинния интеграл от първи род $\iint\limits_S \left(y^2z^2 + z^2x^2 + y^2x^2 \right) ds$. С помощта на въведената в следствие 1 функция $f1$ намираме подинтегралната функция в двойния интеграл от формула (5).

```
In[5]:= x[u_, v_] := u; y[u_, v_] := v; z[u_, v_] := b * Sqrt[u^2 + v^2];
        f[x_, y_, z_] = y^2 * z^2 + x^2 * z^2 + y^2 * x^2;
        f1[f, x, y, z]
```

Out[5]= $\sqrt{1+b^2}\,\left(u^2 v^2 + b^2 \left(u^2+v^2\right)^2\right)$

Понеже областта на интегриране е $D=\left\{(x,y): x^2+y^2-2ax\le 0\right\}$, то функцията *Boole* може да ни помогне при пресмятането на интеграла.

```
In[6]:= Integrate[% * Boole[u^2 + v^2 - 2 * a * u ≤ 0], {u, 0, 2 * a}, {v, -a, a},
         Assumptions → a > 0 && b > 0]
```

Out[6]= $\frac{1}{24}\left(7\,a^6\sqrt{1+b^2}\,\pi + 80\,a^6 b^2\sqrt{1+b^2}\,\pi\right)$

Отговор. $\iint\limits_S \left(y^2z^2+z^2x^2+y^2x^2\right)ds = \frac{1}{24}\left(80b^2+7\right)\pi a^6\sqrt{1+b^2}$.

2) Ще пресметнем повърхнинния интеграл от втори род
$\iint\limits_S y^2z^2dydz + z^2x^2dzdx + y^2x^2dxdy$.

Свеждаме този интеграл към повторен интеграл върху областта
$D=\left\{(x,y): x^2+y^2-2ax\le 0\right\}$ като използваме сумата от равенства (2), (3) и (4).

Тогава подинтегралната функция може да се получи като сума от функциите $f2$, $f3$ и $f4$ определени в следствие 1. Получаваме

```
In[7]:= p[x_, y_, z_] = y^2 * z^2; q[x_, y_, z_] = x^2 * z^2; r[x_, y_, z_] = y^2 * x^2;
        f2[p, x, y, z] + f3[q, x, y, z] + f4[r, x, y, z] // Simplify
```

Out[7]= $u\,v\left(-b^3\,v\sqrt{u^2+v^2} + u\left(v - b^3\sqrt{u^2+v^2}\right)\right)$

Тогава $\iint\limits_S y^2z^2dydz + z^2x^2dzdx + y^2x^2dxdy$ е равен на

```
In[8]:= Integrate[% * Boole[u^2 + v^2 - 2 * a * u ≤ 0], {u, 0, 2 * a}, {v, -a, a},
         Assumptions → a > 0 && b > 0]
```

Out[8]= $\frac{-8192\,a^6 b^3 + 2205\,a^6\,\pi}{7560}$

Пример 6. Нека S е елипсоида $\frac{x^2}{a^2}+\frac{y^2}{b^2}+\frac{z^2}{c^2}=1$, $a>b>c>0$. Ще пресметнем повърхнинния интеграл от първи род $\iint\limits_S \sqrt{\frac{x^2}{a^4}+\frac{y^2}{b^4}+\frac{z^2}{c^4}}\,ds$.

Решение. Записваме повърхнината S параметрично: $x = a\sin(u)\cos(v)$, $y = b\sin(u)\sin(v)$ и $z = c\cos(u)$ за $0 \le u \le \pi$ и $0 \le v \le 2\pi$. Тогава подинтегралната функция в равенство (1) има вида

```
In[9]:= x[u_, v_] := a*Sin[u]*Cos[v]; y[u_, v_] := b*Sin[u]*Sin[v];
        z[u_, v_] := c*Cos[u]; f[x_, y_, z_] = Sqrt[x^2/a^4 + y^2/b^4 + z^2/c^4];
        f1[f, x, y, z]
```

$$\text{Out[9]= } \sqrt{a^2 b^2 \mathrm{Cos}[u]^2 \mathrm{Sin}[u]^2 + c^2 \mathrm{Sin}[u]^4 \left(b^2 \mathrm{Cos}[v]^2 + a^2 \mathrm{Sin}[v]^2\right)} \sqrt{\frac{\mathrm{Cos}[u]^2}{c^2} + \mathrm{Sin}[u]^2 \left(\frac{\mathrm{Cos}[v]^2}{a^2} + \frac{\mathrm{Sin}[v]^2}{b^2}\right)}$$

Тогава $\iint\limits_S \sqrt{\frac{x^2}{a^4} + \frac{y^2}{b^4} + \frac{z^2}{c^4}}\,ds = \iint\limits_D f1\,du\,dv$, където $f1$ е израза получен в Out[9] и $D = \{(u,v): 0 \le u \le \pi, 0 \le v \le 2\pi\}$. Поради симетрията можем да изчислим само интеграла върху парчето от S, което лежи в първи октант и да умножим полученият резултат по осем. Макар и бавно намитаме крайния отговор

```
In[10]:= 8*Integrate[%, {u, 0, Pi/2}, {v, 0, Pi/2}, Assumptions -> a > b > c > 0] //
          Factor
```

$$\text{Out[10]= } \frac{4\left(a^2 b^2 + a^2 c^2 + b^2 c^2\right)\pi}{3\,a\,b\,c}$$

Резултат бихме получили по-бързо, ако забележим, че след като извадим пред първия корен множителя $abs|\sin(u)|$ получаваме точно втория корен

$$\text{In[11]:= } \frac{a^2 b^2 \mathrm{Cos}[u]^2 \mathrm{Sin}[u]^2 + c^2 \mathrm{Sin}[u]^4 \left(b^2 \mathrm{Cos}[v]^2 + a^2 \mathrm{Sin}[v]^2\right)}{a^2 * b^2 * c^2 * \mathrm{Sin}[u]^2} \text{ // Expand}$$

$$\text{Out[11]= } \frac{\mathrm{Cos}[u]^2}{c^2} + \frac{\mathrm{Cos}[v]^2 \mathrm{Sin}[u]^2}{a^2} + \frac{\mathrm{Sin}[u]^2 \mathrm{Sin}[v]^2}{b^2}$$

$$\text{In[12]:= } \frac{\mathrm{Cos}[u]^2}{c^2} + \mathrm{Sin}[u]^2 \left(\frac{\mathrm{Cos}[v]^2}{a^2} + \frac{\mathrm{Sin}[v]^2}{b^2}\right) \text{ // Expand}$$

$$\text{Out[12]= } \frac{\mathrm{Cos}[u]^2}{c^2} + \frac{\mathrm{Cos}[v]^2 \mathrm{Sin}[u]^2}{a^2} + \frac{\mathrm{Sin}[u]^2 \mathrm{Sin}[v]^2}{b^2}$$

Следователно подинтегралната функция е

$$g(u,v) = abs\left|\sin(u)\right|\left(\left(\frac{\cos(u)}{c}\right)^2 + \left(\frac{\cos(v)\sin(u)}{a}\right)^2 + \left(\frac{\sin(v)\sin(u)}{b}\right)^2\right)$$ и интеграла се

изчислява по-бързо

```
In[13]:= g[u_, v_] = a*b*c*Abs[Sin[u]]*(Cos[u]^2/c^2 + Cos[v]^2 Sin[u]^2/a^2 + Sin[u]^2 Sin[v]^2/b^2);
         Integrate[g[u, v], {u, 0, Pi/2}, {v, 0, Pi/2}, Assumptions → a > b > c > 0]
```

$$\text{Out[13]=} \frac{a^2 b^2 \pi + a^2 c^2 \pi + b^2 c^2 \pi}{6 a b c}$$

```
In[14]:= Factor[8*%]
```

$$\text{Out[14]=} \frac{4 (a^2 b^2 + a^2 c^2 + b^2 c^2) \pi}{3 a b c}$$

Пример 7. Нека S е сферата $(x-a)^2 + (y-b)^2 + (z-c)^2 = R^2$, $R > 0$. Ще пресметнем повърхнинния интеграл от втори род $\iint\limits_S x^2 dydz + y^2 dzdx + z^2 dxdy$.

Решение. Записваме повърхнината S параметрично: $x = a + R\sin(u)\cos(v)$, $y = b + R\sin(u)\sin(v)$ и $z = c + R\cos(u)$ за $0 \le u \le \pi$ и $0 \le v \le 2\pi$. Действително

```
In[15]:= x[u_, v_] := a + R*Sin[u]*Cos[v]; y[u_, v_] := b + R*Sin[u]*Sin[v];
         z[u_, v_] := c + R*Cos[u];
         (x[u, v] - a)^2 + (y[u, v] - b)^2 + (z[u, v] - c)^2 // Simplify
```

$\text{Out[15]=} R^2$

Аналогично на пример 5, задача 2, подинтегралната функция в двойния интеграл има вида

```
In[16]:= p[x_, y_, z_] = x^2; q[x_, y_, z_] = y^2; r[x_, y_, z_] = z^2;
         f2[p, x, y, z] + f3[q, x, y, z] + f4[r, x, y, z] // Simplify
```

```
Out[16]= R^2 Sin[u] (Cos[u] (c + R Cos[u])^2 +
            Cos[v] Sin[u] (a + R Cos[v] Sin[u])^2 + Sin[u] Sin[v] (b + R Sin[u] Sin[v])^2)
```

Тогава $\iint\limits_S x^2 dydz + y^2 dzdx + z^2 dxdy$ е равен на

```
In[17]:= Integrate[%, {u, 0, π}, {v, 0, 2*π}, Assumptions → R > 0]
```

$$\text{Out[17]=} \frac{8}{3}(a + b + c)\pi R^3$$

4. *Примери за приложения на повърхнините интеграли във физиката*

4.1. Приложения на повърхнините интеграли от първи род

Нека S е материална повърхнина с повърхнинна плътност $\rho(x,y,z)$ във всяка точка $(x,y,z)\in S$. Ще предполагаме, че написаните по-долу интеграли съществуват. В сила са следните твърдения:

4.1.1. Масата на повърхнината S е равна на $m=\iint\limits_S \rho(x,y,z)\,ds$.

4.1.2. Статистическият момент на S спрямо равнината Oxy е равен на $m_{xy}=\iint\limits_S z\rho(x,y,z)\,ds$. Аналогично статистическият момент - спрямо равнината Ozx е $m_{zx}=\iint\limits_S y\rho(x,y,z)\,ds$, а спрямо равнината Oyz е $m_{yz}=\iint\limits_S x\rho(x,y,z)\,ds$.

4.1.3. Центъра на тежестта на повърхнината S има координати (x_0,y_0,z_0), където

$$x_0=\frac{\iint\limits_S x\rho(x,y,z)\,ds}{\iint\limits_S \rho(x,y,z)\,ds}=\frac{m_{yz}}{m},\quad y_0=\frac{\iint\limits_S y\rho(x,y,z)\,ds}{\iint\limits_S \rho(x,y,z)\,ds}=\frac{m_{zx}}{m},$$

$$z_0=\frac{\iint\limits_S z\rho(x,y,z)\,ds}{\iint\limits_S \rho(x,y,z)\,ds}=\frac{m_{xy}}{m}.$$

4.1.4. Материална точка $M(x_0,y_0,z_0)$ с маса m_0 се привлича от материалната повърхнина S със силата $\vec{F}(x_0,y_0,z_0)=\left(F_x(x_0,y_0,z_0),F_y(x_0,y_0,z_0),F_z(x_0,y_0,z_0)\right)$, където

$$F_x(x_0,y_0,z_0)=\gamma m_0\iint\limits_S \frac{x-x_0}{\|r\|^3}\rho(x,y,z)\,ds,$$

$$F_y(x_0,y_0,z_0)=\gamma m_0\iint\limits_S \frac{y-y_0}{\|r\|^3}\rho(x,y,z)\,ds,$$

$F_z(x_0,y_0,z_0)=\gamma m_0\iint\limits_S \frac{z-z_0}{\|r\|^3}\rho(x,y,z)\,ds$, $r=(x-x_0,y-y_0,z-z_0)$, $\|r\|=\sqrt{r.r}$ и γ е гравитационната константа.

Потенциала на силата на привличане е $W(x_0,y_0,z_0)=\gamma m_0\iint\limits_S \frac{\rho(x,y,z)}{\|r\|}ds$, където, $r=(x-x_0,y-y_0,z-z_0)$, $\|r\|=\sqrt{r.r}$ и γ е гравитационната константа.

4.1.5. Моментите на инерцията на повърхнината S спрямо:
(а) оста Ox е $I_x = \iint\limits_S \left(y^2 + z^2\right)\rho(x,y,z)\,ds$;
(б) равнината Oxy е $I_{xy} = \iint\limits_S z^2 \rho(x,y,z)\,ds$;
(в) началото на координатната система $I_0 = \iint\limits_S \left(x^2 + y^2 + z^2\right)\rho(x,y,z)\,ds$.

Аналогични са формулите спрямо произволна ос, равнина или точка.

Следствие 3. Ще дефинираме функция m, която при дадени параметричен запис на S, $r(u,v) = \left(x(u,v), y(u,v), z(u,v)\right)$ за $(u,v) \in \{(u,v): a \le u \le b, c(u) \le v \le d(u)\}$ и плътност $f(x,y,z)$, $(x,y,z) \in S$ да изчислява масата на S.

Решение. Дефиницията на функцията m представлява запис на формула (1) за интеграла 4.1.1.

```
In[1]:= m[f_, x_, y_, z_, a_, b_, c_, d_, g_] :=
  Module[{n, n1, n2, R, m1}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := √Simplify[n[u, v].n[u, v]] // PowerExpand;
   n1[u_, v_] := f[x[u, v], y[u, v], z[u, v]] * n2[u, v] // Simplify;
   m1 = Integrate[n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g]; m1]
```

Дефинираната функция m зависи от плътността $f(x,y,z)$; от координатните функции $x(u,v), y(u,v), z(u,v)$; от константите a, b; функциите $c(u)$ и $d(u)$ и условията g при които се изчислява интеграла. В тялото на функцията *Module:* n е векторното произведение на r_u с r_v; $n2$ е

$\sqrt{A(u,v)^2 + B(u,v)^2 + C(u,v)^2} = \sqrt{E(u,v)G(u,v) - F^2(u,v)}$.

Ще валидираме дефинираната функция с пример подробно решен в Ф,т.3,стр.281.

Пример 8. Нека S е сфера с радиус r и център $O(0,0,0)$. Ще намерим масата на сферата S, ако:
а) плътността в точка от сферата е равна на разстоянието от точката до оста Oz;
б) плътността в точка от сферата е равна на квадрата на разстоянието от точката до оста Oz.

Решение. а) Записваме плътността с формула $f(x,y,z) = \sqrt{x^2 + y^2}$.

```
In[2]:= f[x_, y_, z_] := √x^2 + y^2
```

Записваме сферата S параметрично с $x = r\sin(u)\cos(v)$, $y = r\sin(u)\sin(v)$ и $z = r\cos(u)$ за $0 \le u \le \pi$ и $0 \le v \le 2\pi$. Действително

```
In[3]:= x[u_, v_] := r*Sin[u]*Cos[v]; y[u_, v_] := r*Sin[u]*Sin[v];
        z[u_, v_] := r*Cos[u]; x[u, v]^2 + y[u, v]^2 + z[u, v]^2 // Simplify
```

Out[3]= r^2

```
In[4]:= ParametricPlot3D[{x[u, v], y[u, v], z[u, v]} /. r → 1, {u, 0, π},
         {v, 0, 2*π}]
```

Out[4]=



Сега остава да запишем условието, че $r > 0$ и да пресметнем масата

```
In[5]:= g = r > 0; m[f, x, y, z, 0, π, 0, 2*π, g]
```

Out[5]= $\pi^2 r^3$

Отговор:Масата е равна на $\pi^2 r^3$.

б) В този случай единствената разлика е че $f(x, y, z) = x^2 + y^2$. За това само предефинираме плътността f и намираме масата

```
In[6]:= f[x_, y_, z_] := x^2 + y^2; m[f, x, y, z, 0, π, 0, 2*π, g]
```

Out[6]= $\frac{8 \pi r^4}{3}$

Отговор:Масата е равна на $\frac{8\pi r^4}{3}$.

Следствие 4. При условията на следствие 3 ще дефинираме функция M , която да изчислява координатите на центъра на тежестта.

Решение. Аналогично на следствие 3, дефиницията на функцията M представлява запис на формулите 4.1.3. като е използвана формула (1).

In[8]:=
```
M[f_, x_, y_, z_, a_, b_, c_, d_, g_] :=
  Module[{n, n1, n2, R, m1, myz, mzx, mxy},
   R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := √(Simplify[n[u, v].n[u, v]]) // PowerExpand;
   n1[u_, v_] := f[x[u, v], y[u, v], z[u, v]] * n2[u, v] // Simplify;
   m1 = Integrate[n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
   myz = Integrate[x[u, v] * n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
   mzx = Integrate[y[u, v] * n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
   mxy = Integrate[z[u, v] * n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
   {myz/m1, mzx/m1, mxy/m1} // Simplify]
```
В тялото на функцията *Module:* n е векторното произведение на r_u с r_v; $n2$ е $\sqrt{A(u,v)^2 + B(u,v)^2 + C(u,v)^2}$; $m1$ е масата на S;
$mxy = \iint_S z\rho(x,y,z)\,ds$, $mzx = \iint_S y\rho(x,y,z)\,ds$, $myz = \iint_S x\rho(x,y,z)\,ds$.

Ще валидираме дефинираната функция с примери подробно решени в Ф,т.3,стр.281.
Пример 9. Нека S е полусфера с радиус r и център $O(0,0,0)$, разположена над равнината Oxy. При условията на пример 8 ще намерим центъра на тежестта на S.
Решение. Както в пример 8 определяме параметричния запис на S.

In[9]:=
```
x[u_, v_] := r * Sin[u] * Cos[v]; y[u_, v_] := r * Sin[u] * Sin[v];
z[u_, v_] := r * Cos[u]
```
In[10]:=
```
ParametricPlot3D[{x[u, v], y[u, v], z[u, v]} /. r → 1, {u, 0, π/2},
  {v, 0, 2 * π}]
```
Out[10]=


Въвеждаме плътността $f(x,y,z)=\sqrt{x^2+y^2}$ и условието $r>0$ и получаваме, че центъра на тежестта има координати

```
In[11]:= f[x_, y_, z_] := Sqrt[x^2 + y^2]; g = r > 0; M[f, x, y, z, 0, π/2, 0, 2*π, g]

Out[11]= {0, 0, 4 r/(3 π)}
```

Ако плътността е $f(x,y,z)=x^2+y^2$, центъра на тежестта има координати

```
In[12]:= f[x_, y_, z_] := x^2 + y^2; M[f, x, y, z, 0, π/2, 0, 2*π, g]

Out[12]= {0, 0, 3 r/8}
```

Пример 10. Нека K е коничната повърхнина с уравнение $z=\frac{h}{r}\sqrt{x^2+y^2}$,
$(x,y)\in\{(x,y):x^2+y^2\le r^2\}$ и плътност $f(x,y,z)=\rho=const$.
а) Ще намерим центъра на тежестта на K.
б) Ще намерим масата на K.

Решение. Параметричния запис на K е $x=u\cos(v)$, $y=u\sin(v)$ и $z=\frac{h}{r}u$ за $0\le u\le r$ и $0\le v\le 2\pi$. Действително

```
In[15]:= x[u_, v_] := u*Cos[v]; y[u_, v_] := u*Sin[v]; z[u_, v_] = h/r*u;
         h/r*Sqrt[(x[u, v]^2 + y[u, v]^2) // Simplify] // PowerExpand

Out[15]= h u/r

In[16]:= ParametricPlot3D[{x[u, v], y[u, v], z[u, v]} /. {r → 1, h → 1}, {u, 0, 1},
          {v, 0, 2*π}, AxesLabel → {x, y, z}]
```

Out[16]=



Сега можем да използваме готовите функции за намитане на центъра на тежестта и масата. Получаваме, че центъра на тежестта има координати

```
In[17]:= f[x_, y_, z_] := ρ; g = r > 0 && h > 0 && ρ > 0;
         M[f, x, y, z, 0, r, 0, 2*π, g]
```

$$\text{Out[17]= } \left\{0,\ 0,\ \frac{2\,h}{3}\right\}$$

Маста на K е равна на

```
In[18]:= m[f, x, y, z, 0, r, 0, 2*π, g]
```

$$\text{Out[18]= } \pi\, r\, \sqrt{h^2 + r^2}\ \rho$$

Забележка 2. При дефиницията на функцията M ние можем да използваме и дефинираната по-рано функция m. Ще дефинираме по този начин функцията $M1$.

```
In[21]:= M1[f_, x_, y_, z_, a_, b_, c_, d_, g_] :=
          Module[{n, n1, n2, R, m1, myz, mzx, mxy},
           R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
           n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
           n2[u_, v_] := √(Simplify[n[u, v].n[u, v]]) // PowerExpand;
           n1[u_, v_] := f[x[u, v], y[u, v], z[u, v]] * n2[u, v] // Simplify;
           m1 = m[f, x, y, z, a, b, c, d, g];
           myz = Integrate[x[u, v] * n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
           mzx = Integrate[y[u, v] * n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
           mxy = Integrate[z[u, v] * n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
           {myz/m1, mzx/m1, mxy/m1} // Simplify]
```

С новата функция отново изчисляваме координатите на центъра на тежестта на K

```
In[22]:= M1[f, x, y, z, 0, r, 0, 2*π, g]
```

$$\text{Out[22]= } \left\{0,\ 0,\ \frac{2\,h}{3}\right\}$$

Следствие 5. При условията на следствие 3 ще дефинираме функция in, която да пресмята моментите на инерцията на S спрямо координатните равнини. Функцията ще зависи от както и по-горе от параметрите на задачата и ще дава вектор с първа координата момента на инерцията на S спрямо координатната равнина Oxy, втора координата - спрямо координатната равнина Oyz и трета координата - спрямо координатната равнина Ozx, т. е. вектора $\left(I_{xy}, I_{yz}, I_{zx}\right)$.

Решение. Дефиницията на функцията in представлява запис на формулите 4.1.5. като е използвана формула (1).

```
In[23]:= in[f_, x_, y_, z_, a_, b_, c_, d_, g_] :=
  Module[{n, n1, n2, R, ixy, iyz, izx}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := √(Simplify[n[u, v].n[u, v]]) // PowerExpand;
   n1[u_, v_] := f[x[u, v], y[u, v], z[u, v]] * n2[u, v] // Simplify;
   ixy = Integrate[z[u, v]^2 * n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
   iyz = Integrate[x[u, v]^2 * n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
   izx = Integrate[y[u, v]^2 * n1[u, v], {u, a, b}, {v, c, d}, Assumptions → g];
   {ixy, iyz, izx} // Simplify]
```

В тялото на функцията *Module* освен използваните в горните следствия означения сме означили $ixy = \iint\limits_S z^2 \rho(x,y,z)\,ds$, $izx = \iint\limits_S y^2 \rho(x,y,z)\,ds$ и $iyz = \iint\limits_S x^2 \rho(x,y,z)\,ds$.

Ще валидираме дефинираната функция с примери подробно решени в Ф,т.3,стр.282.

Пример 11. Нека K е коничната повърхнина с уравнение $z = \frac{h}{r}\sqrt{x^2 + y^2}$, $(x,y) \in \{(x,y) : x^2 + y^2 \le r^2\}$ и плътност $f(x,y,z) = \rho = const$. Ще намерим моментите на инерцията (I_{xy}, I_{yz}, I_{zx}) на S .

Решение. Като използваме параметричното представяне на K от пример 9 получаваме

```
In[24]:= x[u_, v_] := u*Cos[v]; y[u_, v_] := u*Sin[v]; z[u_, v_] = h/r * u;
  f[x_, y_, z_] := ρ; g = r > 0 && h > 0 && ρ > 0;
  in[f, x, y, z, 0, r, 0, 2*π, g]
```

Out[24]= $\left\{\frac{1}{2} h^2 \pi r \sqrt{h^2 + r^2}\ \rho,\ \frac{1}{4} \pi r^3 \sqrt{h^2 + r^2}\ \rho,\ \frac{1}{4} \pi r^3 \sqrt{h^2 + r^2}\ \rho\right\}$

Отговор: $I_{xy} = \frac{1}{2} h^2 \pi r \sqrt{h^2 + r^2}$ и $I_{yz} = I_{zx} = \frac{1}{4} \pi r^3 \sqrt{h^2 + r^2}$.

Следствие 6. При условията на следствие 3 ще дефинираме функция F , която да изчислява силата на привличане на точка $M(x_0, y_0, z_0)$ с маса m_0 от S и потенциала W на силата на привличане.

Решение. Дефиницията на функцията F представлява запис на формулите 4.1.4. като е използвана формула (1).

Функциите F и M ще зависят освен от плътността и параметрите задаващи повърхнината и от координатите на точката M .

а) Дефиниция на функцията F .

```
In[1]:= F[f_, x_, y_, z_, a_, b_, c_, d_, g_, x0_, y0_, z0_] :=
  Module[{n, n1, n2, n3, R, fxyz}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := Sqrt[Simplify[n[u, v].n[u, v] / ((x[u, v] - x0)^2 + (y[u, v] - y0)^2 + (z[u, v] - z0)^2)^3]];
   n1[u_, v_] :=
    f[x[u, v], y[u, v], z[u, v]] * n2[u, v] *
      {(x[u, v] - x0), (y[u, v] - y0), (z[u, v] - z0)} // Simplify;
   fxyz = Integrate[n1[u, v], {u, a, b}, {v, c, d}, Assumptions -> g];
   m0 * fxyz // Simplify]
```

б) Дефиниция на функцията W .

```
In[2]:= W[f_, x_, y_, z_, a_, b_, c_, d_, g_, x0_, y0_, z0_] :=
  Module[{n, n1, n2, n3, R, w}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := Sqrt[Simplify[n[u, v].n[u, v] / ((x[u, v] - x0)^2 + (y[u, v] - y0)^2 + (z[u, v] - z0)^2)]];
   n1[u_, v_] := f[x[u, v], y[u, v], z[u, v]] * n2[u, v] // Simplify;
   w = Integrate[n1[u, v], {u, a, b}, {v, c, d}, Assumptions -> g]; w // Simplify]
```

Пример 12. Нека C е прав кръгов цилиндър с радиус на основата r и височина h .
При условие, че плътността е константата 1 , ще намерим:
а) силата F с която се привлича центъра на основата на C ;
б) потенциала W на повърхнината C в центъра на основата.
Решение. Записваме данните от примера чрез формули. За да запишем уравнението на повърхнината въвеждаме декартовата координатна система така, че центъра на основата на C да съвпада с началото на координатната система, основата на C да лежи върху равнината Oxy и образуващата на C да е успоредна на оста Oz . Тогава C има параметричен запис: $x = r\cos(v)$, $y = r\sin(v)$ и $z = u$ за $0 \le u \le h$ и $0 \le v \le 2\pi$. Действително

```
In[3]:= x[u_, v_] := r * Cos[v]; y[u_, v_] := r * Sin[v]; z[u_, v_] := u;
  ParametricPlot3D[{x[u, v], y[u, v], z[u, v]} /. r -> 1, {u, 0, e}, {v, 0, 2 * π},
   AxesLabel -> {x, y, z}]
```

Out[3]=



Сега остава да въведем плътността $f(x, y, z) = 1$ и условията $r > 0$ и $h > 0$. Тогава дефинираната функция F в следствие 6 дава търсената сила на привличане.

```
In[4]:= f[x_, y_, z_] := 1; g = r > 0 && h > 0;
        F[f, x, y, z, 0, h, 0, 2*π, g, 0, 0, 0]
```

Out[4]= $\left\{0, 0, \pi \left(2 - \frac{2\, r}{\sqrt{h^2 + r^2}}\right) m_0\right\}$

При така въведените параметри на задачата потенциала на повърхнината в центъра на основата се дава от функцията W.

```
In[5]:= W[f, x, y, z, 0, h, 0, 2*π, g, 0, 0, 0]
```

Out[5]= $2\,\pi\, r\, \mathrm{ArcSinh}\left[\frac{h}{r}\right]$

Сравняваме полученият резултат в Ф с 282. Виждаме, че отговорите съвпадат, защото $\operatorname{arcsinh}(t) = \ln\left(t + \sqrt{1 + t^2}\right)$. Действително

In[6]:= p[t_] = ArcSinh[t] - Log$\left[t + \sqrt{1 + t^2}\right]$; D[p[t], t] // FullSimplify

Out[6]= 0

и $p(t) = \operatorname{arcsinh}(t) - \ln\left(t + \sqrt{1 + t^2}\right) = const.$ Освен това

```
In[7]:= p[0]
```

Out[7]= 0

Пример 13. Нека K е конуса от пример 10.

А) При условие, че плътността е константата 1, ще намерим:
а) потенциала W на повърхнината K в центъра на основата и неговия връх;
б) силата F с която се привлича центъра на основата на K.
Б) При условие, че плътността е равна на разстоянието от точката до върха на конуса, ще намерим:
а) потенциала W на повърхнината K в неговия връх;
б) силата F с която се привлича върха на K.
Решение. А) Въвеждаме функциите x, y, z, които задават конуса. Въвеждаме плътността $f = 1$ и условията $r > 0$ и $h > 0$.
а) Понеже центъра на основата има координати $(0, 0, h)$, то потенциала W е равен на

```
In[8]:= x[u_, v_] := u*Cos[v]; y[u_, v_] := u*Sin[v]; z[u_, v_] := h/r*u;
        f[x_, y_, z_] := 1; g = r > 0 && h > 0;
        W[f, x, y, z, 0, r, 0, 2*π, g, 0, 0, h]
```

Out[8]= $$\frac{2\pi r\left((-h+r)\sqrt{h^2+r^2}+h^2\,\mathrm{Log}\left[\frac{r\left(r+\sqrt{h^2+r^2}\right)}{h\left(-h+\sqrt{h^2+r^2}\right)}\right]\right)}{h^2+r^2}$$

Записваме отговора и по друг начин

```
In[9]:= Expand[%]
```

Out[9]= $$-\frac{2h\pi r}{\sqrt{h^2+r^2}}+\frac{2\pi r^2}{\sqrt{h^2+r^2}}+\frac{2h^2\pi r\,\mathrm{Log}\left[\frac{r\left(r+\sqrt{h^2+r^2}\right)}{h\left(-h+\sqrt{h^2+r^2}\right)}\right]}{h^2+r^2}$$

Понеже върхът на конуса има координати $(0, 0, 0)$, то потенциала W е равен на

```
In[10]:= W[f, x, y, z, 0, r, 0, 2*π, g, 0, 0, 0]
```

Out[10]= $2\pi r$

б) Изчисляваме силата F с която се привлича центъра на основата на K

```
In[11]:= F[f, x, y, z, 0, r, 0, 2*π, g, 0, 0, h]
```

Out[11]= $$\left\{0,\ 0,\ -\frac{2\pi\left((h+r)\sqrt{h^2+r^2}+h\,r\,\mathrm{Log}\left[\frac{h\left(-h+\sqrt{h^2+r^2}\right)}{r\left(r+\sqrt{h^2+r^2}\right)}\right]\right)m_0}{h^2+r^2}\right\}$$

In[12]:= Expand[%]

Out[12]= $\left\{0,0,-\frac{2h\pi m_0}{\sqrt{h^2+r^2}}-\frac{2\pi r m_0}{\sqrt{h^2+r^2}}-\frac{2h\pi r \operatorname{Log}\left[\frac{h\left(-h+\sqrt{h^2+r^2}\right)}{r\left(r+\sqrt{h^2+r^2}\right)}\right] m_0}{h^2+r^2}\right\}$

Б) а)Предефинираме плътността така, че $f(x,y,z)$ да е равна на разстоянието от (x,y,z) до върха $(0,0,0)$ и намираме потенциала W в точката $(0,0,0)$

```
In[13]:= f[x_, y_, z_] := √(x^2 + y^2 + z^2) ;
         W[f, x, y, z, 0, r, 0, 2*π, g, 0, 0, 0]
```

Out[13]= $\pi r \sqrt{h^2+r^2}$

б)Изчисляваме силата F с която се привлича върха на K .

```
In[14]:= F[f, x, y, z, 0, r, 0, 2*Pi, g, 0, 0, 0]
```

Out[14]= $\left\{0,0,\frac{2h\pi r m_0}{\sqrt{h^2+r^2}}\right\}$

Забележка 3. Често при изчисляването на интегралите е удобно използването на полярна смяна на променливите. Ето как изглеждат функциите F и W , ако повърхнината е графиката на функцията $z=z(x,y)$, $(x,y)\in D$ и след полярна смяна на координатите D се записва като $\Omega=\left((r,\varphi):a\le\varphi\le b,c(\varphi)\le r\le d(\varphi)\right)$

```
In[15]:= Fzp[f_, z_, a_, b_, c_, d_, g_, x0_, y0_, z0_] :=
          Module[{n1, n2, n3, fxyz},
           n2[x_, y_] := √(Factor[1 + (D[z[x, y], x])^2 + (D[z[x, y], y])^2]) //
             PowerExpand;
           n1[x_, y_] =
            {x - x0, y - y0, z[x, y] - z0} * f[x, y, z[x, y]] *
                                n2[x, y]
             ---------------------------------------------------- // Simplify;
             ((x - x0)^2 + (y - y0)^2 + (z[x, y] - z0)^2)^(3/2)
           fxyz = Integrate[n1[u*Cos[v], u*Sin[v]]*u, {v, a, b}, {u, c, d},
             Assumptions -> g]; m0 * fxyz // Simplify]
```

```
In[16]:= Wzp[f_, z_, a_, b_, c_, d_, g_, x0_, y0_, z0_] :=
          Module[{n1, n2, n3, w},
            n2[x_, y_] := √(Factor[1 + (D[z[x, y], x])^2 + (D[z[x, y], y])^2]) //
              PowerExpand;
            n1[x_, y_] = f[x, y, z[x, y]] * n2[x, y] / ((x - x0)^2 + (y - y0)^2 + (z[x, y] - z0)^2)^(1/2) //
              Simplify;
            w = Integrate[n1[u * Cos[v], u * Sin[v]] * u, {v, a, b}, {u, c, d}, Assumptions → g];
            m0 * w // Simplify]
```

Решете пример 12 с помощта на функциите Fzp и Wzp.

4.2. Приложения на повърхнините интеграли от втори род

Нека S е двустранна повърхнина и $n(x,y,z)$ е избраното векторно поле от нормали. Това за произволно непрекъснато върху S векторно поле $v(x,y,z)$ с пълния интеграл от втори род $\iint\limits_S (v,n)\,ds$ се определя потока на векторното поле $v(x,y,z)$ преминаващ през S.

Пример 14. Нека върху повърхнината E_1 определена с $\frac{x^2}{a^2}+\frac{y^2}{b^2}+\frac{z^2}{c^2}=1$, $z \ge 0$ е избрана външната нормала. Ще определим потока на векторното поле $v=\left(x^3, yz, z\right)$.

Решение. Записваме E_1 параметрично: $x = a\sin u\cos v$, $y = b\sin u\sin v$ и $z = c\cos u$ за $0 \le u \le \frac{\pi}{2}$ и $0 \le v \le 2\pi$. Действително

```
In[1]:= x[u_, v_] := a * Sin[u] * Cos[v]; y[u_, v_] := b * Sin[u] * Sin[v];
        z[u_, v_] := c * Cos[u];

In[2]:= x[u, v]^2/a^2 + y[u, v]^2/b^2 + z[u, v]^2/c^2 // Simplify

Out[2]= 1
```

Скицираме повърхнината E_1

```
In[3]:= ParametricPlot3D[{x[u, v], y[u, v], z[u, v]} /. {a → π, b → e, c → 1}, {u, 0, π/2},
          {v, 0, 2 * π}, AxesLabel → {x, y, z}]
```

Out[3]=



Дефинираме функцията P_0, която изчислява потока на векторно поле $v(x,y,z)=\left(p(x,y,z),q(x,y,z),r(x,y,z)\right)$ преминаващ през повърхнина зададена параметрично $x=x(u,v)$, $y=y(u,v)$ и $z=z(u,v)$ за $v\in[a,b]$ и $u\in[c,d]$ при условия g.

```
In[4]:= P0[p_, q_, r_, x_, y_, z_, a_, b_, c_, d_, g_] :=
  Module[{n, n1, V, R, po}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   V[u_, v_] := {p[x[u, v], y[u, v], z[u, v]], q[x[u, v], y[u, v], z[u, v]],
     r[x[u, v], y[u, v], z[u, v]]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n1[u_, v_] := V[u, v].n[u, v] // Simplify;
   po := Integrate[n1[u, v], {v, a, b}, {u, c, d}, Assumptions → g]; po // Simplify]
```

Понеже вече са въведени функциите $x=a\sin u\cos v$, $y=b\sin u\sin v$ и $z=c\cos u$, въвеждаме векторното поле и условията g. Изчисляваме потока през повърхнината E_1 с функцията P_0.

```
In[5]:= p[x_, y_, z_] := x^3; q[x_, y_, z_] := y*z; r[x_, y_, z_] := z; g = a > 0 && b > 0 && c > 0;
  P0[p, q, r, x, y, z, 0, 2*Pi, 0, π/2, g]

Out[5]= 1/60 a b c (40 + 24 a^2 + 15 c) π
```

Отговор. Потока е равен на $\frac{1}{60}abc\left(40+24a^2+15c\right)$.

Пример 15. Нека S е повърхнината на тялото $T=\left\{(x,y,z),x^2+y^2+z^2\le 3r^2,0\le z\le\sqrt{x^2+y^2-r^2}\right\}$ и $v=\left(x^2,-y^2,z^2\right)$ е векторно поле. Ще изчислим потока на v през външната повърхнина на T.

Решение. Представяме си тялото T като положим $r=\pi$ и използваме функцията *RegionPlot3D.*

```
In[2]:= RegionPlot3D[0 ≤ z ≤ Sqrt[x^2 + y^2 - π^2] && x^2 + y^2 + z^2 ≤ 3*π^2, {x, -Sqrt[3]*π, Sqrt[3]*π},
  {y, -Sqrt[3]*π, Sqrt[3]*π}, {z, 0, Sqrt[2]*π}, Mesh → None, PlotPoints → 50,
  PlotStyle → FaceForm[Yellow, Blue], AxesLabel → {"x", "y", "z"}]
```



Забелязваме, че неговата граница се състои от три части. Първата част означаваме с S_1 и тя е вътрешната част, която е оцветена в син цвят. Частта S_1 е част от отсечения конус $z = \sqrt{x^2 + y^2 - r^2}$

```
In[2]:= Plot3D[√(x² + y² - π²), {x, -√3 * π, √3 * π}, {y, -√3 * π, √3 * π},
   RegionFunction -> Function[{x, y, z}, x² + y² + z^2 ≤ 3 * π²],
   AxesLabel → {"x", "y", "z"}]
```



Понеже външната нормала за границата на тялото T е вътрешната нормала за коничната повърхнина, то тя може да се изчисли по този начин

```
In[3]:= x[u_, v_] := u; y[u_, v_] := v; z[u_, v_] := Sqrt[u^2 + v^2 - r^2];
        R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
        n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
        n[u, v]
```

Out[3]= $\left\{-\frac{u}{\sqrt{-r^2+u^2+v^2}}, -\frac{v}{\sqrt{-r^2+u^2+v^2}}, 1\right\}$

Втората част означаваме с S_2 и е частта на полукълбото $z=\sqrt{3r^2-x^2-y^2}$, която е извън отсечения конус

```
In[4]:= Plot3D[Sqrt[-x^2 - y^2 + 3*π^2], {x, -Sqrt[3]*π, Sqrt[3]*π}, {y, -Sqrt[3]*π, Sqrt[3]*π},
         PlotStyle → FaceForm[Yellow],
         RegionFunction -> Function[{x, y, z}, 0 ≤ z ≤ Sqrt[x^2 + y^2 - π^2]], Mesh → None,
         AxesLabel → {"x", "y", "z"}]
```

Out[4]=



Понеже външната нормала за границата на тялото T е външна нормала и за полусферата, то тя може да се изчисли по този начин

```
In[5]:= z[u_, v_] := Sqrt[-u^2 - v^2 + 3*r^2]; n[u, v]
```

Out[5]= $\left\{\frac{u}{\sqrt{3r^2-u^2-v^2}}, \frac{v}{\sqrt{3r^2-u^2-v^2}}, 1\right\}$

Третата част на S означаваме с S_3. Тя е основата на тялото и е венеца $\left\{(x,y,z): r^2\le x^2+y^2\le 3r^2, z=0\right\}$. Частта S_3 се записва параметрично така:

```
In[6]:= x[u_, v_] := u*Cos[v]; y[u_, v_] := u*Sin[v]; z[u_, v_] := 0;
        ParametricPlot3D[{x[u, v], y[u, v], z[u, v]}, {u, π, Sqrt[3]*π}, {v, 0, 2*π},
         AxesLabel → {x, y, z}]
```

Out[6]=



Нормала към S_3 е равна на

```
In[7]:= n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
        -n[u, v]

Out[7]= {0, 0, -u}
```

Повърхнините S_2 и S_3 се пресичат в линията $\gamma:\begin{cases} z=\sqrt{3r^2-x^2-y^2} \\ z=\sqrt{x^2+y^2-r^2} \end{cases}$, т.е.

$x^2+y^2=2r^2$ и $z=r$. Следователно S_1 се задава като графика на функцията

$z=\sqrt{x^2+y^2-r^2}$ за $(x,y)\in D_1=\left\{(x,y): r^2\le x^2+y^2\le 2r^2\right\}$, а S_2 се задава като графика на функцията

$z=\sqrt{3r^2-x^2-y^2}$ за $(x,y)\in D_2=\left\{(x,y): 2r^2\le x^2+y^2\le 3r^2\right\}$. Границите на областите D_1 и D_2 са концентрични окръжности и можем да ги скицираме

```
In[8]:= ContourPlot[{x^2 + y^2 == 3 * π^2, x^2 + y^2 == 2 * π^2, x^2 + y^2 == π^2},
          {x, -√3 * π, √3 * π}, {y, -√3 * π, √3 * π}]
```

Out[8]=



Записваме подинтегралната функция при свеждане пълния повърхнинен интеграл от втори ред към повторен интеграл, т. е. функцията

$$f_0(u,v) = P(x(u,v), y(u,v), z(u,v))A(u,v) + Q(x(u,v), y(u,v), z(u,v))B(u,v) +$$
$$+R(x(u,v), y(u,v), z(u,v))C(u,v)$$

```
In[9]:= f0[p_, q_, r_, x_, y_, z_] :=
         Module[{n, n1, n2, R, V}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
          V[u_, v_] := {p[x[u, v], y[u, v], z[u, v]], q[x[u, v], y[u, v], z[u, v]],
            r[x[u, v], y[u, v], z[u, v]]};
          n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
          n2[u_, v_] := V[u, v].n[u, v]; n1[u_, v_] := n2[u, v] // Simplify; n1[u, v]]
```

Понеже $\iint_S (v,n)\,ds = \iint_{S_1} (v,n)\,ds + \iint_{S_2} (v,n)\,ds + \iint_{S_3} (v,n)\,ds$, отначало изчисляваме $A = \iint_{S_1} (v,n)\,ds$. Намираме подинтегралната функция при свеждането на интеграла A към повторен интеграл

```
In[10]:= p[x_, y_, z_] := x^2; q[x_, y_, z_] := -y^2; r[x_, y_, z_] := z^2;
         x[u_, v_] := u; y[u_, v_] := v; z[u_, v_] := Sqrt[u^2 + v^2 - r^2];
         f0[p, q, r, x, y, z]
```

$$\text{Out[10]=}\quad -r^2 + u^2 + v^2 - \frac{u^3}{\sqrt{-r^2 + u^2 + v^2}} + \frac{v^3}{\sqrt{-r^2 + u^2 + v^2}}$$

Тази функция трябва да се интегрира по венеца D_1. Извършваме полярна смяна на променливите $u = s\cos(t)$, $v = s\sin(t)$. Тогава получената в Out[10] функция има вида

```
In[11]:= % /. {u -> s*Cos[t], v -> s*Sin[t]} // Simplify
```

$$\text{Out[11]=}\quad -\frac{4 r^2 \sqrt{-r^2 + s^2} - 4 s^2 \sqrt{-r^2 + s^2} + 3 s^3 \operatorname{Cos}[t] + s^3 \operatorname{Cos}[3 t] - 3 s^3 \operatorname{Sin}[t] + s^3 \operatorname{Sin}[3 t]}{4 \sqrt{-r^2 + s^2}}$$

В новите променливи областта D_1 се записва $\left\{(s,t) : r \le s \le \sqrt{2}r, 0 \le t \le 2\pi\right\}$. Понеже якобияна на смяната е равен на s, то търсеният интеграл е равен на

```
In[12]:= A = Integrate[s*%, {t, 0, 2*Pi}, {s, r, Sqrt[2]*r}, Assumptions -> r > 0]
```

$$\text{Out[12]=}\quad \frac{\pi r^4}{2}$$

По аналогичен начин изчисляваме $B = \iint_{S_2} (v,n)\,ds$. Намираме подинтегралната функция при свеждането на интеграла B към повторен интеграл

```
In[13]:= z[u_, v_] := Sqrt[-u^2 - v^2 + 3*r^2]; f0[p, q, r, x, y, z]
```

$$\text{Out[13]=}\quad 3r^2 - u^2 + \frac{u^3}{\sqrt{3r^2-u^2-v^2}} + v^2\left(-1 - \frac{v}{\sqrt{3r^2-u^2-v^2}}\right)$$

Тази функция трябва да се интегрира по венеца D_2. Извършваме полярна смяна на променливите $u = s\cos(t)$, $v = s\sin(t)$ и изчисляваме B

```
In[14]:= % /. {u -> s*Cos[t], v -> s*Sin[t]} // Simplify
```

$$\text{Out[14]=}\quad \frac{12r^2\sqrt{3r^2-s^2} - 4s^2\sqrt{3r^2-s^2} + 3s^3\,\text{Cos}[t] + s^3\,\text{Cos}[3t] - 3s^3\,\text{Sin}[t] + s^3\,\text{Sin}[3t]}{4\sqrt{3r^2-s^2}}$$

```
In[15]:= B = Integrate[s*%, {t, 0, 2*Pi}, {s, Sqrt[2]*r, Sqrt[3]*r}, Assumptions -> r > 0]
```

$$\text{Out[15]=}\quad \frac{\pi r^4}{2}$$

Намираме подинтегралната функция при свеждането на интеграла $\iint_{S_3} (v,n)\,ds$ към повторен интеграл

```
In[16]:= p[x_, y_, z_] := x^2; q[x_, y_, z_] := -y^2; r[x_, y_, z_] := z^2;
         x[u_, v_] := u*Cos[v]; y[u_, v_] := u*Sin[v]; z[u_, v_] := 0;
         f0[p, q, r, x, y, z]
```

Out[16]= 0

Понеже е равна на нула, то и интеграла е нула. Следователно
$\iint_S (v,n)\,ds = \iint_{S_1} (v,n)\,ds + \iint_{S_2} (v,n)\,ds + \iint_{S_3} (v,n)\,ds$ е равен на

```
In[17]:= B + A // Simplify
```

$$\text{Out[17]=}\quad \pi r^4$$

5. *Основни интегрални формули*

Ще бъдат представени две основни формули, които често се използват.

Теорема 2 (Гаус) . Нека G е ограничена област в R^3, чиято граница S е сума от краен брой частично гладки повърхнини. Нека освен това векторното поле $v(x,y,z) = \big(P(x,y,z), Q(x,y,z), R(x,y,z)\big)$ има непрекъснати частни производни

$\frac{\partial P(x,y,z)}{\partial x}$, $\frac{\partial Q(x,y,z)}{\partial y}$ и $\frac{\partial R(x,y,z)}{\partial z}$ върху $\bar{G}$. Означаваме с S^+ границата S с избирана ориентацията задавана от външната нормала. Тогава е в сила формулата

$$\iiint_G \left(\frac{\partial P(x,y,z)}{\partial x} + \frac{\partial Q(x,y,z)}{\partial y} + \frac{\partial R(x,y,z)}{\partial z} \right) dxdydz = \iint_{S^+} Pdydx + Qdzdx + Rdxdy .$$

Това равенство се записва съкратено така

$$\iiint_G \operatorname{div}(v)\, dxdydz = \iint_S v.\vec{dS} ,$$

където са използвани стандартните означения

$\operatorname{div}(v) = \frac{\partial P(x,y,z)}{\partial x} + \frac{\partial Q(x,y,z)}{\partial y} + \frac{\partial R(x,y,z)}{\partial z}$, $\vec{dS} = (\cos\alpha, \cos\beta, \cos\gamma)ds$ и $(\cos\alpha, \cos\beta, \cos\gamma)$ е векторното поле на единичните външни нормали.

Формулата на Гаус е известна и като формула на Гаус-Остроградски или като теорема за дивергенцията. Тя остава в сила и когато интегралите се разбират като несобствени.

Първо ще разгледаме два примера на приложение на формулата на Гаус, когато повърхнинния интеграл се свежда до пресмятане на интеграл от дивергенцията по тялото оградено от повърхнината.

Пример 16. Ще пресметнем повърхнинния интеграл от втори род $\iint_S x^2dydz + y^2dzdx + z^2dxdy$, където S е сферата $(x-a)^2 + (y-b)^2 + (z-c)^2 = R^2$, $R > 0$ с ориентация определена от външната нормала, като използваме формулата на Гаус. Полученият резултат ще сравним с пример 7.

Решение. Означаваме с K кълбото $\left\{(x,y,z) : (x-a)^2 + (y-b)^2 + (z-c)^2 \le R^2\right\}$.

Според формулата на Гаус е изпълнено равенството

$\iint_S x^2dydz + y^2dzdx + z^2dxdy = \iiint_K \left(\frac{\partial x^2}{\partial x} + \frac{\partial y^2}{\partial y} + \frac{\partial z^2}{\partial z} \right) dxdydz$. Пресмятаме дивергенцията

```
In[1]:= f[x_, y_, z_] = D[x^2, x] + D[y^2, y] + D[z^2, z]

Out[1]= 2 x + 2 y + 2 z
```

Понеже интеграла е върху кълбо, то е удобно да се направи сферична смяна на променливите $x = a + t\sin(u)\cos(v)$, $y = b + t\sin(u)\sin(v)$, $z = c + t\cos(u)$ за $0 \le t \le r$, $0 \le u \le \pi$ и $0 \le v \le 2\pi$.

```
In[2]:= x[t_, u_, v_] := a + t*Sin[u]*Cos[v]; y[t_, u_, v_] := b + t*Sin[u]*Sin[v];
        z[t_, u_, v_] := c + t*Cos[u];
        (x[t, u, v] - a)^2 + (y[t, u, v] - b)^2 + (z[t, u, v] - c)^2 ≤ r^2 // Simplify

Out[2]= t² ≤ r²
```

Изчисляваме якобияна на смяната на променливите

```
In[3]:= j[t_, u_, v_] = Det[D[{x[t, u, v], y[t, u, v], z[t, u, v]}, {{t, u, v}}]] //
          FullSimplify

Out[3]= t² Sin[u]
```

Тогава пространствения интеграл в сферичните променливи е равен

```
In[4]:= Integrate[f[x[t, u, v], y[t, u, v], z[t, u, v]]*j[t, u, v], {t, 0, r},
          {u, 0, π}, {v, 0, 2*π}]

        8
Out[4]= - (a + b + c) π r³
        3
```

Полученият резултат ще съвпада с отговора от пример 7.

Пример 17. Нека S е коничната повърхнина с уравнение $z = \frac{h}{r}\sqrt{x^2+y^2}$, $(x,y) \in \left\{(x,y) : x^2+y^2 \le r^2\right\}$ и ориентация определена от външната нормала. Ще пресметнем повърхнинния интеграл от втори род $\iint\limits_S x^2dydz + y^2dzdx + z^2dxdy$.

Решение. Повърхнината е скицирана в пример 10. Означаваме с S_1 основата на конуса $\left\{(x,y,z) : z = h, x^2+y^2 \le r^2\right\}$ с ориентация определена от горната нормала, т.е. вектора $(0,0,1)$. Тялото K с граница $S \cup S_1$ е конус. От формулата на Гаус следва

$$\iint\limits_{S\cup S_1} x^2dydz + y^2dzdx + z^2dxdy = \iiint\limits_K \left(\frac{\partial x^2}{\partial x} + \frac{\partial y^2}{\partial y} + \frac{\partial z^2}{\partial z}\right)dxdydz ,$$

т. е.

$$\iint\limits_S x^2dydz + y^2dzdx + z^2dxdy = \iiint\limits_K \left(\frac{\partial x^2}{\partial x} + \frac{\partial y^2}{\partial y} + \frac{\partial z^2}{\partial z}\right)dxdydz - \iint\limits_{S_1} x^2dydz + y^2dzdx + z^2dxdy .$$

Лесно се забелязва, че $B = \iint\limits_{S_1} x^2dydz + y^2dzdx + z^2dxdy = h^2\int\limits_{S_1} ds = h^2\pi r^2$. Ние ще получим резултата като използваме функцията $f0$ от пример 15.

```
In[1]:= f0[p_, q_, r_, x_, y_, z_] :=
  Module[{n, n1, n2, R, V}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   V[u_, v_] := {p[x[u, v], y[u, v], z[u, v]], q[x[u, v], y[u, v], z[u, v]],
     r[x[u, v], y[u, v], z[u, v]]};
   n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := V[u, v].n[u, v]; n1[u_, v_] := n2[u, v] // Simplify; n1[u, v]]
```

Записваме S_1 параметрично и изчисляваме интеграла B.

```
In[2]:= x[u_, v_] := u*Cos[v]; y[u_, v_] := u*Sin[v]; z[u_, v_] := h;
 p[x_, y_, z_] := x^2; q[x_, y_, z_] := y^2; r[x_, y_, z_] := z^2;
 f0[p, q, r, x, y, z]

Out[2]= h^2 u

In[3]:= B = Integrate[%, {v, 0, 2*π}, {u, 0, r}, Assumptions -> h > 0 && r > 0]

Out[3]= h^2 π r^2
```

За да намерим интеграла $A=\iiint\limits_K \left(\frac{\partial x^2}{\partial x}+\frac{\partial y^2}{\partial y}+\frac{\partial z^2}{\partial z}\right)dxdydz$, първо пресмятаме дивергенцията

```
In[4]:= f[x_, y_, z_] = D[x^2, x] + D[y^2, y] + D[z^2, z]

Out[4]= 2 x + 2 y + 2 z
```

Понеже интеграла е върху конуса K, то извършваме цилиндрична смяна на променливите $x=t\cos(v)$, $y=t\sin(v)$ и $z=u$. Въвеждаме смяната на променливите

```
In[5]:= x[t_, u_, v_] := t*Cos[v]; y[t_, u_, v_] := t*Sin[v]; z[t_, u_, v_] := u;
 Simplify[h/r * Sqrt[x[t, u, v]^2 + y[t, u, v]^2]] // PowerExpand

Out[5]= h t / r
```

Понеже $0 \le v \le 2\pi$, то за да определим границите в които варират другите променливи, разглеждаме сечението на K с равнината Oyz

Out[6]=



Вижда се, че ако $u \in [0, h]$, то $t \in \left[0, \dfrac{u.r}{h}\right]$, защото триъгълника $\triangle OUD$ е подобен на триъгълника $\triangle OAB$.

Въвеждаме новите променливи и пресмятаме якобияна на смяната на променливите

```
In[7]:= Clear[x, y, z, u, v]

In[8]:= x[t_, u_, v_] := t*Cos[v]; y[t_, u_, v_] := t*Sin[v]; z[t_, u_, v_] := u;
        j[t_, u_, v_] = Det[D[{x[t, u, v], y[t, u, v], z[t, u, v]}, {{t, v, u}}]] //
          FullSimplify

Out[8]= t
```

Тогава за интеграла A получаваме

```
In[9]:= A = Integrate[f[x[t, u, v], y[t, u, v], z[t, u, v]] * j[t, u, v], {v, 0, 2*π},
          {u, 0, h}, {t, 0, u*r/h}, Assumptions → h > 0 && r > 0]
```

Out[9]= $\frac{1}{2} h^2 \pi r^2$

Следователно интеграла $\iint\limits_S x^2 dydz + y^2 dzdx + z^2 dxdy$ е равен на

```
In[10]:= A - B
```

Out[10]= $-\frac{1}{2} h^2 \pi r^2$

Формулата на Гаус често се прилага и с цел да се пресметне интеграл върху тримерно тяло с интеграл върху неговата граница. Например пресмятането на обем се свежда до пресмятане на повърхнинен интеграл от втори род чрез следствие 6.

Следствие 6. 1) Обема V на ограничена област T чиято граница S е сума от краен брой частично гладки повърхнини може да се изчисли с формулата

$$V = \frac{1}{3}\iint\limits_{S^+} xdydz + ydzdx + zdxdy = \iint\limits_{S^+} xdydz = \iint\limits_{S^+} ydzdx = \iint\limits_{S^+} zdxdy .$$

2) Ако повърхнината S се задава параметрично с гладките функции $x = x(u,v)$, $y = y(u,v)$, $z = z(u,v)$ за $(u,v) \in D$, то $V = \frac{1}{3}\iint\limits_{D} \det\begin{pmatrix} x & y & z \\ x'_u & y'_u & z'_u \\ x'_v & y'_v & z'_v \end{pmatrix} dudv$

Пример 17. В случая когато границата S на тялото T се задава параметрично с гладките функции $x = x(u,v)$, $y = y(u,v)$, $z = z(u,v)$ за $(u,v) \in \{(u,v) : a \le v \le b, c(v) \le u \le d(v)\}$ ще определим функция която изчислява обема на V .

Решение.

```
In[16]:= Clear[V, x, y, z, u, v, g, a, b, c, d]

In[17]:= V[x_, y_, z_, a_, b_, c_, d_, g_] :=
          1/3 * Module[{n, R, vo}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
            n[u_, v_] := Det[{R[u, v], D[R[u, v], u], D[R[u, v], v]}] // Simplify;
            vo := Integrate[n[u, v], {v, a, b}, {u, c, d}, Assumptions -> g]; vo // Simplify]
```

С помощта на дефинираната функция пресмятаме обема на елипсоида $E : \frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2} \le 1$. За целта записваме неговата граница параметрично $x = a\sin(u)\cos(v)$, $y = a\sin(u)\sin(v)$, $x = c\cos(u)$, за $0 \le u \le \pi$ и $0 \le v \le 2\pi$. Тогава обема е равен на

```
In[18]:= x[u_, v_] := a * Sin[u] * Cos[v]; y[u_, v_] := b * Sin[u] * Sin[v];
         z[u_, v_] := c * Cos[u]; g := a > 0 && b > 0 && c > 0;
         V[x, y, z, 0, 2 * Pi, 0, π, g]

Out[18]= 4/3 a b c π
```

Следствие 7. Обема на конус ограден от гладка конична повърхнина и основа S лежаща на произволна равнина е равен на $V = \frac{|S|h}{3}$, където $|S|$ е лицето на основата S и h е височината на конуса.

Пример 18. Нека S_0 е повърхнина зададена параметрично с $x = a\cos(u)\cos(v) + b\sin(u)\sin(v)$, $y = a\cos(u)\sin(v) - b\sin(u)\cos(v)$, $z = c\sin(u)$

за $u \in \left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$ и $v \in [0, 2\pi]$. Предполагаме, че $c > 0$, $a > 0$ и $b > 0$. Ще намерим обема на тялото T с околна повърхнина S_0.

Решение. Въвеждаме функциите x, y, z и скицираме повърхнината в частния случаи $c = 4$, $a = 3$ и $b = 1$ и когато $c = 4$, $a = 1$ и $b = 3$.

In[1]:=
```
x[u_, v_] := a*Cos[u]*Cos[v] + b*Sin[u]*Sin[v];
y[u_, v_] := a*Cos[u]*Sin[v] - b*Sin[u]*Cos[v]; z[u_, v_] := c*Sin[u];
ParametricPlot3D[{x[u, v], y[u, v], z[u, v]} /. {a → 3, b → 1, c → 4},
 {u, -π/2, π/2}, {v, 0, 2*π}, AxesLabel → {x, y, z}]
```

Out[1]=



In[2]:=
```
ParametricPlot3D[{x[u, v], y[u, v], z[u, v]} /. {a → 1, b → 3, c → 4},
 {u, -π/2, π/2}, {v, 0, 2*π}, AxesLabel → {x, y, z}]
```



И в двата случая забелязваме, че тялото T има долна основа S_1 при $z=-c$, т.е. при $u=-\dfrac{\pi}{2}$ и горна основа S_2 при $z=c$, т.е. при $u=\dfrac{\pi}{2}$. Границата на долната основа S_1 има параметричен запис

In[3]:= $\left\{x\left[-\frac{\pi}{2}, v\right], y\left[-\frac{\pi}{2}, v\right], z\left[-\frac{\pi}{2}, v\right]\right\}$ // Simplify

Out[3]= {-b Sin[v], b Cos[v], -c}

и очевидно е окръжност с радиус b лежаща в равнината $z=-c$ с център $(0,0,-c)$.

In[4]:= $x\left[-\frac{\pi}{2}, v\right]$^2 + $y\left[-\frac{\pi}{2}, v\right]$^2 // Simplify

Out[4]= b^2

Повърхнината S_1 е кръга с радиус b лежаща в равнината $z=-c$. Границата на горната основа S_2 има параметричен запис

$\left\{x\left[\frac{\pi}{2}, v\right], y\left[\frac{\pi}{2}, v\right], z\left[\frac{\pi}{2}, v\right]\right\}$ // Simplify

{b Sin[v], -b Cos[v], c}

и очевидно е окръжност с радиус b лежаща в равнината $z=c$ с център $(0,0,c)$.

In[5]:= $x\left[\frac{\pi}{2}, v\right]$^2 + $y\left[\frac{\pi}{2}, v\right]$^2 // Simplify

Out[5]= b^2

Повърхнината S_2 е кръга с радиус b лежаща в равнината $z=c$.

```
In[6]:= {ParametricPlot[{-b Sin[v], b Cos[v]} /. b → 3, {v, 0, 2 π}, AxesLabel → {x, y}],
        ParametricPlot[{b Sin[v], -b Cos[v]} /. b → 3, {v, 0, 2 π}, AxesLabel → {x, y}]}
```

Out[6]= 

Ще намерим обема на T , като използваме следствие 6, т.е.

$V = \dfrac{1}{3}\iint\limits_{S^+} xdydx + ydzdx + zdxdy$, където S^+ е $S_0^+ \cup S_1^+ \cup S_2^+$ с избрана външна нормала.

Ще използваме факта, че $\iint\limits_{S^+} xdydx + ydzdx + zdxdy = A + B_1 + B_2$, където

$A = \iint\limits_{S_0^+} xdydx + ydzdx + zdxdy$, $B_1 = \iint\limits_{S_1^+} xdydx + ydzdx + zdxdy$ и

$B_2 = \iint\limits_{S_2^+} xdydx + ydzdx + zdxdy$.

За да намерим интеграла A , пресмятаме нормалния вектор към S , като използваме параметричния запис на повърхнината.

```
In[7]:= R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
        n[u_, v_] := Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
        n[u, v]

Out[7]= {-c Cos[u] (a Cos[u] Cos[v] + b Sin[u] Sin[v]),
         c Cos[u] (b Cos[v] Sin[u] - a Cos[u] Sin[v]), (-a^2 + b^2) Cos[u] Sin[u]}
```

Забелязваме, че той има вида

$\left(-cx(u,v)\cos(u), -cy(u,v)\cos(u), (b^2 - a^2)\sin(u)\cos(u)\right)$ и е насочен навътре в тялото T . Следователно при свеждането на интеграла A към повторен, аналогично на пример 15, определяме функцията изчисляваща подинтегралната функция

```
In[8]:= f01[p_, q_, r_, x_, y_, z_] :=
  Module[{n, n1, n2, R, V}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   V[u_, v_] := {p[x[u, v], y[u, v], z[u, v]], q[x[u, v], y[u, v], z[u, v]],
     r[x[u, v], y[u, v], z[u, v]]};
   n[u_, v_] := -1*Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := V[u, v].n[u, v]; n1[u_, v_] := n2[u, v] // Simplify; n1[u, v]];
 p[x_, y_, z_] := x; q[x_, y_, z_] := y; r[x_, y_, z_] := z;
 f01[p, q, r, x, y, z]

Out[8]= a^2 c Cos[u]
```

Разликата с функцията $f0$ от пример 15 е само в дефиницията на нормалния вектор $n(u,v)$. Тогава A е равно на

```
In[9]:= A = Integrate[%, {v, 0, 2*Pi}, {u, -π/2, π/2}]

Out[9]= 4 a^2 c π
```

Пресмятането на $B_1 = \iint\limits_{S_1^+} xdydx + ydzdx + zdxdy$ може да стане на ръка поради тривиалността на изчисленията. Ние обаче ще зададем S_1 параметрично с равенствата $x = u\cos(v)$, $y = u\sin(v)$, $z = -c$ за $v \in [0, 2\pi]$ и ползвайки определената функция $f01$ изчисляваме B_1.

```
In[10]:= x[u_, v_] := u*Cos[v]; y[u_, v_] := u*Sin[v]; z[u_, v_] := -c;
 f01[p, q, r, x, y, z]

Out[10]= c u

In[11]:= B1 = Integrate[%, {v, 0, 2*π}, {u, 0, b}]

Out[11]= b^2 c π
```

Аналогично намираме и числото B_2

```
In[12]:= f0[p_, q_, r_, x_, y_, z_] :=
  Module[{n, n1, n2, R, V}, R[u_, v_] := {x[u, v], y[u, v], z[u, v]};
   V[u_, v_] := {p[x[u, v], y[u, v], z[u, v]], q[x[u, v], y[u, v], z[u, v]],
     r[x[u, v], y[u, v], z[u, v]]};
   n[u_, v_] := -1*Cross[D[R[u, v], u], D[R[u, v], v]] // Simplify;
   n2[u_, v_] := V[u, v].n[u, v]; n1[u_, v_] := n2[u, v] // Simplify; n1[u, v]];
 B2 = Integrate[f0[p, q, r, x, y, z], {v, 0, 2*π}, {u, 0, b}]

Out[12]= b^2 c π
```

Следователно обема е равен на

In[13]:= $\frac{1}{3}*(A+B1+B2)$

Out[13]= $\frac{1}{3}\left(4a^2c\pi+2b^2c\pi\right)$

Теорема 3. (Стокс) Нека S^+ е с частично гладка, двустранна повърхнина с избрана ориентация. Предполагаме, че границата γ^+ на S е частично гладка с ориентация индуцирана от избраната ориентация на S. Ако векторното поле $v(x,y,z)=\left(P(x,y,z),Q(x,y,z),R(x,y,z)\right)$ има непрекъснати частни производни в околност на S, то е в сила формулата

$$\iint\limits_{S^+}\left(\frac{\partial R}{\partial y}-\frac{\partial Q}{\partial z}\right)dydz+\left(\frac{\partial P}{\partial z}-\frac{\partial R}{\partial x}\right)dzdx+\left(\frac{\partial Q}{\partial x}-\frac{\partial P}{\partial y}\right)dxdy=\oint\limits_{\gamma^+}Pdx+Qdy+Rdz\ .$$

Векторното поле $\left(\frac{\partial R}{\partial y}-\frac{\partial Q}{\partial z},\frac{\partial P}{\partial z}-\frac{\partial R}{\partial x},\frac{\partial Q}{\partial x}-\frac{\partial P}{\partial y}\right)$ се нарича ***ротор*** на векторното поле $v(x,y,z)=\left(P(x,y,z),Q(x,y,z),R(x,y,z)\right)$ и се означава с $\operatorname{rot}(v)$. Възприет е и следния запис на ротора $\operatorname{rot}(v)=\det\begin{pmatrix} i & j & k \\ \frac{\partial}{\partial x} & \frac{\partial}{\partial y} & \frac{\partial}{\partial z} \\ P & Q & R\end{pmatrix}$.

Формулата на Стокс може да се запише съкратено така

$$\iint\limits_{S^+}\operatorname{rot}(v)\,\vec{dS}=\oint\limits_{\gamma^+}v\,\vec{dr}\ ,$$

където $\vec{dS}=(\cos\alpha,\cos\beta,\cos\gamma)ds$, $(\cos\alpha,\cos\beta,\cos\gamma)$ е векторното поле на единичните нормали на S^+ и $\vec{dr}=(dx,dy,dz)$.

Забележка 4. Формулата на Стокс остава в сила и когато γ^+ е обединение от краен брой частично гладки свързани компоненти. Тогава интеграла $\oint\limits_{\gamma^+}v\,\vec{dr}$ е сума от криволинейните интеграли по свързаните компоненти.

Областта $G\subset R^3$ се нарича ***повърхнинно едносвързана*** когато за произволна затворена начупена линия γ, изцяло съдържаща се в G, може да се намери частично гладка повърхнина, също принадлежаща на G, за която γ е граница.

Теорема 4. Нека векторното поле $v(x,y,z)=\left(P(x,y,z),Q(x,y,z),R(x,y,z)\right)$ е непрекъснато в областта G. Тогава следните три твърдения са еквивалентни.

1. За произволна, затворен, частично гладка линия γ съдържаща се в G е изпълнено $\oint\limits_{\gamma} Pdx + Qdy + Rdz = 0$

2. За произволни точки A и B от G е в сила равенството
$\int\limits_{\gamma_1} Pdx + Qdy + Rdz = \int\limits_{\gamma_2} Pdx + Qdy + Rdz$,

където γ_i са произволни частично гладки линии принадлежаща на G с начало точката A и край точката B.

3. Съществува такава непрекъснато диференцируема функция $U(x, y, z)$ в G, че са в сила равенствата $\frac{\partial U}{\partial x} = P$, $\frac{\partial U}{\partial y} = Q$ и $\frac{\partial U}{\partial z} = R$.

Функцията $U(x, y, z)$, която удовлетворява твърдението 3 се нарича ***потенциал на векторното поле*** v.

Следствие 8. При условие, че е в сила поне едно от твърденията на теорема 4, то са в сила следните твърдения:

1) Ако $U(x, y, z)$ е потенциал, частично гладката линия γ принадлежи на G, има начало точката A и край точката B, то
$\int\limits_{\gamma} Pdx + Qdy + Rdz = U(B) - U(A)$.

2) Фиксираме точката $A(x_0, y_0, z_0) \in G$. Определяме функцията
$U(x, y, z) = \int\limits_{\gamma} Pdx + Qdy + Rdz$, където $B(x, y, z) \in G$ и $\gamma \subset G$ е частично гладка линия с начало A, край B. Тогава $U(x, y, z)$ е потенциал на v в дефиниционната си област.

3) Ако точката $A(x_0, y_0, z_0) \in G$, всяка точката $B(x, y, z)$ от отвореното подмножество Ω на G начупената линия AA_1A_2B, където $A_1(x, y_0, z_0)$, $A(x, y, z_0)$ изцяло се съдържа в G, то функцията

$$U(x, y, z) = \int\limits_{x_0}^{x} P(t, y_0, z_0)\, dt + \int\limits_{y_0}^{y} Q(x, t, z_0)\, dt + \int\limits_{z_0}^{z} R(x, y, t)\, dt$$

е потенциал на полето v.

Теорема 5. Нека векторното поле $v(x, y, z) = (P(x, y, z), Q(x, y, z), R(x, y, z))$ има непрекъснати частни производни от първи ред в ***повърхнинно едносвързаната*** област G. Тогава всяко едно от твърденията 1, 2 и 3 от теорема 4 е еквивалентно на следното твърдение 4.

4. В областта G са налице равенствата $\frac{\partial R}{\partial y} = \frac{\partial Q}{\partial z}$, $\frac{\partial P}{\partial z} = \frac{\partial R}{\partial x}$ и $\frac{\partial Q}{\partial x} = \frac{\partial P}{\partial y}$.

Пример 19. 1) Ще дефинираме функцията rot, която изчислява ротора на дадено векторно поле.
2) Ще дефинираме функцията PL, която изчислява потенциала на дадено векторно поле в условията на твърдение 3 от следствие 8.
3) С така дефинираните функции ще покажем, че векторното поле $v = (y+z, x+z, x+y)$ е потенциално и ще намерим един негов потенциал.

Решение.1) Дефинираме функцията rot отново с помощта на функцията *Module*.

```
In[1]:= rot[p_, q_, r_] :=
  Module[{R},
   R[x_, y_, z_] :=
    {∂y r[x, y, z] - ∂z q[x, y, z], ∂z p[x, y, z] - ∂x r[x, y, z],
     ∂x q[x, y, z] - ∂y p[x, y, z]} // Simplify; R[x, y, z] // FullSimplify]
```

2) Функцията PL ще зависи от векторното поле $v = (p, q, r)$, от координатите на точките $A(x_0, y_0, z_0)$ и $B(x, y, z)$.

```
In[2]:= PL[p_, q_, r_, x0_, y0_, z0_, x_, y_, z_] :=
  Module[{R1, R2, R3, R}, R1 := Integrate[p[t, y0, z0], {t, x0, x}] // Simplify;
   R2 := Integrate[q[x, t, z0], {t, y0, y}] // Simplify;
   R3 := Integrate[r[x, y, t], {t, z0, z}] // Simplify; R := R1 + R2 + R3; R // Simplify]
```

3) Въвеждаме координатните функции на векторното поле $v = (y+z, x+z, x+y)$ и с дефинираната функция rot изчисляваме ротора

```
In[3]:= p[x_, y_, z_] := y + z; q[x_, y_, z_] := x + z; r[x_, y_, z_] := x + y;
  rot[p, q, r]

Out[3]= {0, 0, 0}
```

Следователно векторното поле $v = (y+z, x+z, x+y)$ е потенциално и може с определената по-горе функция PL да намерим един потенциал.

```
In[4]:= PL[p, q, r, 0, 0, 0, x, y, z]

Out[4]= y z + x (y + z)
```

Проверка.

```
In[5]:= {D[%, x], D[%, y], D[%, z]} - {p[x, y, z], q[x, y, z], r[x, y, z]} // Simplify

Out[5]= {0, 0, 0}
```

Намерената функция е потенциал, защото нейния градиент съвпада с векторното поле v.

Пример 20. Нека линията γ има параметричен запис $x = 1 - 2\sin^2(t)$,
$y = \sin(t) - 3\cos(t)$, $z = \sin(t) - \cos(t)$ за $t \in \left[0, \frac{\pi}{2}\right]$ с посока определена от нарастването на параметъра t. Ще пресметнем интеграла
$J = \int_\gamma \left(yxe^x + ze^y + ye^z\right)dx + \left(xze^y + ze^x + xe^z\right)dy + \left(xye^z + ye^x + xe^y\right)dz$.

Решение. С помощта на определената в предния пример функция rot проверяваме дали векторното поле $v = \left(\left(yxe^x + ze^y + ye^z\right), \left(xze^y + ze^x + xe^z\right), \left(xye^z + ye^x + xe^y\right)\right)$ е потенциално.

```
In[6]:= p[x_, y_, z_] := y*z*Exp[x] + z*Exp[y] + y*Exp[z];
        q[x_, y_, z_] := x*z*Exp[y] + z*Exp[x] + x*Exp[z];
        r[x_, y_, z_] := x*y*Exp[z] + y*Exp[x] + x*Exp[y];
        rot[p, q, r]

Out[6]= {0, 0, 0}
```

Следователно v е потенциално. Сега с определената в пример 19 функция PL намираме един потенциал на полето

```
In[7]:= PL[p, q, r, 0, 0, 0, x, y, z]

Out[7]= ⅇ^z x y + ⅇ^y x z + ⅇ^x y z
```

Проверяваме дали намерената функция е потенциал

```
In[8]:= {D[%, x], D[%, y], D[%, z]} - {p[x, y, z], q[x, y, z], r[x, y, z]} // Simplify

Out[8]= {0, 0, 0}
```

Кривата γ е гладка. Ще изчислим интеграла J с помощта на следствие 8. За целта намираме началото A на γ.

```
In[9]:= s[t_] = {1 - 2*Sin[t]^2, Sin[t] - 3*Cos[t], Sin[t] - Cos[t]}; s[0]

Out[9]= {1, -3, -1}
```

Намираме и края B на γ.

```
In[10]:= s[π/2]

Out[10]= {-1, 1, 1}
```

Според следствие 8, твърдение 1, интеграла J е равен на

```
In[11]:= PL[p, q, r, 0, 0, 0, -1, 1, 1] - PL[p, q, r, 0, 0, 0, 1, -3, -1]
```

Out[11]= $\frac{1}{e^3} + \frac{4}{e} - 5e$